
Preface

The SMT-LIB initiative is an international effort, supported by several research groups world-
wide, with the two-fold goal of producing an extensive on-line library of benchmarks and
promoting the adoption of common languages and interfaces for SMT solvers. This document
specifies Version 2.6 of the SMT-LIB Standard which is a backward-compatible extension of
Version 2.5.

CHAPTER 1

General Information

1.1 About This Document

This document is mostly self-contained, though it assumes some familiarity with first-order
logic, aka predicate calculus. The reader is referred to any of several textbooks on the to-
pic [Gal86, Fit96, End01, Men09]. Previous knowledge of Version 1.2 of the SMT-LIB stan-
dard [RT06] is not necessary. In fact, Version 1.2 users are warned that this version, while
largely based on Version 1.2, is not backward compatible with it. See the Version 2.0 docu-
ment [BST10b] for a summary of the major differences.

This document provides BNF-style abstract and concrete syntax for a number of SMT-LIB
languages. Only the concrete syntax is part of the official SMT-LIB standard. The abstract syntax is
used here mainly for descriptive convenience; adherence to it is not prescribed. Implementors
are free to use whatever internal structure they please for their abstract syntax trees.

New releases of the document are identified by their release date. Each new release of the
same version of the SMT-LIB standard contains, by and large, only conservative additions and
changes with respect to the standard described in the previous release, as well as improve-
ments to the presentation. The only non-conservative changes may be error fixes.

Historical notes and explanations of the rationale of design decisions in the definition of
the SMT-LIB standard are provided in Appendix A, with reference in the main text given as a
superscript number enclosed in parentheses.

1.1.1 Change log for Version 2.6

Release: 2017-07-18

• First release of Version 2.6.

1.1.2 Differences between Version 2.6 and Version 2.5

The SMT-LIB 2 language and logic now supports user-defined algebraic datatypes. Such types
can be introduced with two new commands: declare-datatype, to declare a single algebraic
datatype, and declare-datatypes, to declare two or more mutually recursive datatypes (see
Section 4.2.3). The language of terms now includes a new binder, match, for pattern-matching-
based case analysis over datatype values (see Section 3.6.1). The variables bound by match are
those that occur in patterns, also a new addition to the language. The underlying logic has also
been modified to provide built-in semantics for the constructor, selector and tester symbols
defined with each new algebraic datatype (see Chapter 5).

This version also makes official a number of set-info attributes used in benchmarks from
the official SMT-LIB repository and specifies some requirements on their occurrence and order
(see Section 3.9.3 and Section 4.2.9). Any other differences in this document are only edits
to improve the presentation. Except for the addition of algebraic datatypes, which is fully
backward compatible, the rest of the SMT-LIB language is unchanged.

1.1.3 Change log for Version 2.5

Release: 2015-06-28

• Clarified in Section 4.1.1 that the exit command can be issued in any mode.

Release: 2015-05-28

• First release of Version 2.5.

1.1.4 Differences between Version 2.5 and Version 2.0

Version 2.5 is an extension of Version 2.0 and, with two minor exceptions, is fully backward
compatible with it. There is then no need to have separate support for 2.0 if one supports
Version 2.6. The following list summarizes notable differences and extensions. The first two
items are the only non-backward compatible changes.

• There is now a different set of escape sequences for string literals. It consists of a single
sequence, "", used to represent the double quote character within the literal.

• The predefined option :expand-definitions has been removed because there are now
no cases in which it applies.

• SMT-LIB source files are not limited to the US-ASCII format anymore and can now
consist of Unicode characters. The concrete encoding is currently left unspecified, but
should be a compatible 8-bit extension of the 7-bit US-ASCII set, such as UTF-8.

• We have clarified several points about the character set used by the SMT-LIB language
and specified more precisely which characters are allowed in string literals, identifiers
and symbols.

• We have made explicit several details on the scoping and shadowing rules for identifiers,
in particulars those occurring in binders.

• Identifiers can now be indexed not just with numerals but also with symbols.

• The use of the term attribute :pattern and its related syntax for quantifier patterns has
been made official.

• The solver option :interactive-mode has been renamed :produce-assertions. The old
name is still accepted but its use is now deprecated.

• There is now a predefined argument, ALL, for the set-logic command which refers to
the most general logic supported by the solver executing the command.

• We have introduced a notion of execution mode for a solver to better describe the restric-
tion on when commands can be executed or options set.

• There is a new solver option :global-declarations that makes all definitions and decla-
rations global and not removable by pop operations. Global declarations can be removed
only by the new command reset.

• The new command reset brings the state of a solver to the state it had immediately
after start up (resetting everything).

• The new command reset-assertions empties the assertion stack and removes all as-
sertions. If :global-declarations is set to false, it also removes all declarations and
definitions.

• The new command check-sat-assuming checks the satisfiability of the current context
under an additional number of assumptions provided as input to the command. When it
returns unsat, a new companion command, get-unsat-assumptions, returns the subset
of input assumptions used by the solver to prove the context unsatisfiable. The latter
command is enabled or disabled with the new option :produce-unsat-assumptions.
The old check-sat command can now be defined, conservatively, as a special case of
check-sat-assuming with an empty set of assumptions.

• The new command declare-const can now be used to declare nullary function symbols.

• The new command echo prints back on the regular output channel a string provided as
input.

• The new commands define-fun-rec and define-funs-rec respectively allow the defi-
nition of recursive functions and of sets of mutually recursive functions.

• The new command get-model returns a representation of a model computed by the
solver in response to an invocation of the check-sat or check-sat-assuming command.

• The new get-info flag :assertion-stack-levels returns the current number of levels
in the assertion stack.

• The new option :reproducible-resource-limit can be used to set a solver-defined
resource limit that applies to each invocation of check-sat or check-sat-assumptions.

1.1.5 Typographical and notational conventions

The concrete syntax of the SMT-LIB language is defined by means of BNF-style production
rules. In the concrete syntax notation, terminals are written in typewriter font, as in false,
while syntactic categories (non-terminals) are written in slanted font and enclosed in angular
brackets, as in 〈term〉. In the production rules, the meta-operators ::= and | are used as usual
in BNF. Also, as usual, the meta-operators _∗ and _+ denote zero, respectively, one, or more
repetitions of their argument. We use _n and _n+1 instead of _∗ and _+ when we want to
indicate that multiple occurrences to the latter operators have the same number of repetitions.
We use the notation ddec (resp., ehex) to represent the Unicode character with decimal code d
(resp., hexadecimal code e). Remember that the US-ASCII character with code d < 128 is also
the Unicode character ddec. Examples of concrete syntax expressions are provided in shaded
boxes like the following.

(f (- x) x)

In the abstract syntax notation, which uses the same meta-operators as the concrete syntax,
words in boldface as well as the symbols ≈, ∃, ∀, and Π denote terminal symbols, while words
in italics and Greek letters denote syntactic categories. For instance, x, σ are non-terminals and
Bool is a terminal. Parentheses are meta-symbols, used just for grouping—they are not part
of the abstract language. Function applications are denoted simply by juxtaposition, which is
enough at the abstract level.

To simplify the notation, when there is no risk of confusion, the name of an abstract
syntactic category is also used, possibly with subscripts, to denote individual elements of that
category. For instance, t is the category of terms and t (as well as t1, t2 and so on) is also used
to denote individual terms.

The meta-syntax x̄ denotes a sequence of the form x1x2 · · · xn for some x1, x2, . . . , xn and
n ≥ 0.

1.2 Overview of SMT-LIB

Satisfiability Modulo Theories (SMT) is an area of automated deduction that studies methods
for checking the satisfiability of first-order formulas with respect to some logical theory T
of interest [BSST09]. What distinguishes SMT from general automated deduction is that the
background theory T need not be finitely or even first-order axiomatizable, and that special-
ized inference methods are used for each theory. By being theory-specific and restricting their
language to certain classes of formulas (such as, typically but not exclusively, quantifier-free
formulas), these specialized methods can be implemented in solvers that are more efficient in
practice than general-purpose theorem provers.

While SMT techniques have been traditionally used to support deductive software ver-
ification, they have found applications in other areas of computer science such as, for in-

http://www.utf8-chartable.de/unicode-utf8-table.pl?utf8=dec

stance, planning, model checking and automated test generation. Typical theories of interest
in these applications include formalizations of various forms of arithmetic, arrays, finite sets,
bit vectors, algebraic datatypes, strings, floating point numbers, equality with uninterpreted
functions, and various combinations of these.

1.2.1 What is SMT-LIB?

SMT-LIB is an international initiative, coordinated by the authors of this document and en-
dorsed by a large number of research groups world-wide, aimed at facilitating research and
development in SMT [BST10a]. Since its inception in 2003, the initiative has pursued these
aims by focusing on the following concrete goals: provide standard rigorous descriptions of
background theories used in SMT systems; develop and promote common input and output
languages for SMT solvers; establish and make available to the research community a large
library of benchmarks for SMT solvers.

The main motivation of the SMT-LIB initiative was the expectation that the availability of
common standards and of a library of benchmarks would greatly facilitate the evaluation and
the comparison of SMT systems, and advance the state of the art in the field, in the same
way as, for instance, the TPTP library [Sut09] has done for theorem proving, or the SATLIB
library [HS00] did for propositional satisfiability. These expectations have been largely met,
thanks in no small part to extensive benchmark contributions from the research community
and to an annual SMT solver competition, SMT-COMP [BdMS05], based on benchmarks from
the library.

At the time of this writing, the library contains more than 100,000 benchmarks and contin-
ues to grow. Formulas in SMT-LIB format are accepted by the great majority of current SMT
solvers. Moreover, much published experimental work in SMT relies significantly on SMT-LIB
benchmarks.

1.2.2 Main features of the SMT-LIB standard

The previous main version of the SMT-LIB standard, Version 1.2, provided a language for
specifying theories, logics (see later), and benchmarks, where a benchmark was, in essence, a
logical formula to be checked for satisfiability with respect to some theory.

Version 2.0 sought to improve the usefulness of the SMT-LIB standard by simplifying its
logical language while increasing its expressiveness and flexibility. In addition, it introduced
a command language for SMT solvers that expanded their SMT-LIB interface considerably,
allowing users to tap the numerous functionalities that most modern SMT solvers provide.

Like Version 2.0 and later versions, Version 2.6 defines:

• a language for writing terms and formulas in a sorted (i.e., typed) version of first-order
logic;

• a language for specifying background theories and fixing a standard vocabulary of sort,
function, and predicate symbols for them;

• a language for specifying logics, suitably restricted classes of formulas to be checked for
satisfiability with respect to a specific background theory;

• a command language for interacting with SMT solvers via a textual interface that allows
asserting and retracting formulas, querying about their satisfiability, examining their
models or their unsatisfiability proofs, and so on.



CHAPTER 2

Basic Assumptions and Structure

This chapter introduces the defining basic assumptions of the SMT-LIB standard and describes
its overall structure.

2.1 Satisfiability Modulo Theories

The defining problem of Satisfiability Modulo Theories is checking whether a given (closed)
logical formula ϕ is satisfiable, not in general but in the context of some background theory T
which constrains the interpretation of the symbols used in ϕ. Technically, the SMT problem
for ϕ and T is the question of whether there is a model of T that makes ϕ true.

A dual version of the SMT problem, which we could call Validity Modulo Theories, asks
whether a formula ϕ is valid in some theory T , that is, satisfied by every model of T . As the
name suggests, SMT-LIB focuses only on the SMT problem. However, at least for classes of
formulas that are closed under logical negation, this is no restriction because the two prob-
lems are inter-reducible: a formula ϕ is valid in a theory T exactly when its negation is not
satisfiable in the theory.

Informally speaking, SMT-LIB calls an SMT solver any software system that implements a
procedure for satisfiability modulo some given theory. In general, one can distinguish among
a solver’s

1. underlying logic, e.g., first-order, modal, temporal, second-order, and so on,

2. background theory, the theory against which satisfiability is checked,

3. input formulas, the class of formulas the solver accepts as input, and

4. interface, the set of functionalities provided by the solver.

For instance, in a solver for linear arithmetic the underlying logic is first-order logic with
equality, the background theory is the theory of real numbers, and the input language may
be limited to conjunctions of inequations between linear polynomials. The interface may be
as simple as accepting a system of inequations and returning a binary response indicating
whether the system is satisfiable or not. More sophisticated interfaces include the ability
to return concrete solutions for satisfiable inputs, return proofs for unsatisfiable ones, allow
incremental and backtrackable input, and so on.

For better clarity and modularity, the aspects above are kept separate in SMT-LIB. SMT-
LIB’s commitment to each of them is described in the following.

2.2 Underlying Logic

Version 2.6 of the SMT-LIB format adopts as its underlying logic a version of many-sorted
first-order logic with equality [Man93, Gal86, End01]. Like traditional many-sorted logic, it
has sorts (i.e., basic types) and sorted terms. Unlike that logic, however, it does not have a
syntactic category of formulas distinct from terms. Formulas are just sorted terms of a dis-
tinguished Boolean sort, which is interpreted as a two-element set in every SMT-LIB theory.1

Furthermore, the SMT-LIB logic uses a language of sort terms, as opposed to just sort con-
stants, to denote sorts: sorts can be denoted by sort constants like Int as well as sort terms like
(List (Array Int Real)). Finally, in addition to the usual existential and universal quanti-
fiers, the logic includes a let binder and a match binder analogous to constructs with the same
name found in functional programming languages.

SMT-LIB’s underlying logic, henceforth SMT-LIB logic, provides the formal foundations of
the SMT-LIB standard. The concrete syntax of the logic is part of the SMT-LIB language of
formulas and theories, which is defined in Part II of this document. An abstract syntax for
SMT-LIB logic and the logic’s formal semantics are provided in Part III.

2.3 Background Theories

One of the goals of the SMT-LIB initiative is to clearly define a catalog of background theories,
starting with a small number of popular ones, and adding new ones as solvers for them are
developed.2 Theories are specified in SMT-LIB independently of any benchmarks or solvers.
On the other hand, each SMT-LIB script refers, indirectly, to one or more theories in the SMT-
LIB catalog.

This version of the SMT-LIB standard distinguishes between basic theories and combined
theories. Basic theories, such as the theory of real numbers, the theory of arrays, the theory
of fixed-size bit vectors and so on, are those explicitly defined in the SMT-LIB catalog. Com-
bined theories are defined implicitly in terms of basic theories by means of a general modular
combination operator. The difference between a basic theory and a combined one in SMT-LIB
is essentially operational. Some SMT-LIB theories, such as the theory of finite sets with a

1This is similar to some formulations of classical higher-order logic, such as that of [And86].
2 This catalog is available, separately from this document, from the SMT-LIB website (www.smt-lib.org).

http://www.smt-lib.org

cardinality operator, are defined as basic theories, even if they are in fact a combination of
smaller theories, because they cannot be obtained by modular combination.

Theory specifications have mostly documentation purposes. They are meant to be standard
references for human readers. For practicality then, the format insists that only the signature
of a theory (essentially, its set of sort symbols and sorted function symbols) be specified
formally—provided it is finite.3 By “formally” here we mean written in a machine-readable
and processable format, as opposed to written in free text, no matter how rigorously. By this
definition, theories themselves are defined informally, in natural language. Some theories,
such as the theory of bit vectors, have an infinite signature. For them, the signature too is
specified informally in English.(1)

2.4 Input Formulas

SMT-LIB adopts a single and general first-order (sorted) language in which to write logical
formulas. It is often the case, however, that SMT applications work with formulas expressed
in some particular fragment of the language. The fragment in question matters because one
can often write a solver specialized on that sublanguage that is much more efficient than a
solver meant for a larger sublanguage.4

An extreme case of this situation occurs when satisfiability modulo a given theory T is
decidable for a certain fragment (quantifier-free, say) but undecidable for a larger one (full
first-order, say), as for instance happens with the theory of arrays [BMS06]. But a similar
situation occurs even when the decidability of the satisfiability problem is preserved across
various fragments. For instance, if T is the theory of real numbers, the satisfiability in T of
full-first order formulas built with the symbols {0, 1,+, ∗,<,=} is decidable. However, one
can implement increasingly faster solvers by restricting the language respectively to quantifier-
free formulas, linear equations and inequations, difference inequations (inequations of the
form x < y + n), and inequations between variables [BBC+05].

Certain pairs of theories and input languages are very common in the field and are often
conveniently considered as a single entity. In recognition of this practice, the SMT-LIB format
allows one to pair together a background theory and an input language into a sublogic, or,
more briefly, logic. We call these pairs (sub)logics because, intuitively, each of them defines a
sublogic of SMT-LIB logic for restricting both the set of allowed models—to the models of the
background theory—and the set of allowed formulas—to the formulas in the input language.

2.5 Interface

Starting with Version 2.0, the SMT-LIB standard includes a scripting language that defines a
textual interface for SMT solvers. SMT solvers implementing this interface act as interpreters
of the scripting language. The language is command-based, and defines a number of in-
put/output functionalities that go well beyond simply checking the satisfiability of an input

3 The finiteness condition can be relaxed a bit for signatures that include certain commonly used sets of
constants such as the set of all numerals.

4 By efficiency here we do not necessarily refer to worst-case time complexity, but efficiency in practice.

formula. It includes commands for setting various solver parameters, declaring new sym-
bols, asserting and retracting formulas, checking the satisfiability of the current set of asserted
formulas, inquiring about models of satisfiable sets, printing various diagnostics, and so on.

CHAPTER 3

The SMT-LIB Language

This chapter defines and explains the concrete syntax of the SMT-LIB standard, what we
comprehensively refer to as the SMT-LIB language. The SMT-LIB language has three main
components: theory declarations, logic declarations, and scripts. Its syntax is similar to that of the
LISP programming language. In fact, every expression in this version is a legal S-expression of
Common Lisp [Ste90]. The choice of the S-expression syntax and the design of the concrete
syntax was mostly driven by the goal of simplifying parsing, as opposed to facilitating human
readability.(2)

The three main components of the language are defined in this chapter by means of BNF-
style production rules. The rules, with additional details, are also provided in Appendix B.
The language generated by these rules is actually a superset of the SMT-LIB language. The
legal expressions of the language must satisfy additional constraints, such as well-sortedness,
also specified in this document.

3.1 Lexicon

The syntax rules in this chapter are given directly with respect to streams of lexical tokens
from the set defined in this section. The whole set of concrete syntax rules is also available for
easy reference in Appendix B.

SMT-LIB source files consist of Unicode characters in any 8-bit encoding, such as UTF-8,
that extends the original 7-bit US-ASCII set. While not technically Unicode, the ISO 8859-1
character set is also allowed since it coincides with the first 256 characters of UTF-8.(3)

Most lexical tokens defined below are limited to US-ASCII printable characters, namely,
characters 32dec to 126dec. The remaining printable characters, mostly used for non-English
alphabet letters (characters 128dec and beyond), are allowed in string literals, quoted symbols,
and comments.

http://en.wikipedia.org/wiki/UTF-8
http://en.wikipedia.org/wiki/ISO_8859-1
http://en.wikipedia.org/wiki/ASCII#ASCII_printable_characters

A comment is any character sequence not contained within a string literal or a quoted sym-
bol (see later) that begins with the semi-colon character ; and ends with the first subsequent
line-breaking character, i.e., 10dec or 13dec. Both comments and consecutive white space char-
acters occurring outside a string literal or a symbol (see later) are considered whitespace. The
only lexical function of whitespace is to break the source text into tokens.1

The lexical tokens of the language are the parenthesis characters ( and ), the elements
of the syntactic categories 〈numeral〉, 〈decimal〉, 〈hexadecimal〉, 〈binary〉, 〈string〉, 〈symbol〉,
〈keyword〉, as well as a number of reserved words, all defined below together with a few auxil-
iary syntactic categories.

White Space Characters. A 〈white_space_char〉 is one of the following characters: 9dec (tab),
10dec (line feed), 13dec (carriage return), and 32dec (space).

Printable Characters. A 〈printable_char〉 is any character from 32dec to 126dec (US-ASCII) and
from 128dec on.2

Digits. A 〈digit〉 is any character from 48dec to 57dec (0 through 9)

Letters. A 〈letter〉 is any character from 65dec to 90dec (English alphabet letters A through Z)
and from 97dec to 122dec (English alphabet letters a through z).(4)

Numerals. A 〈numeral〉 is the digit 0 or a non-empty sequence of digits not starting with 0 .

Decimals. A 〈decimal〉 is a token of the form 〈numeral〉.0∗〈numeral〉 .

Hexadecimals. A 〈hexadecimal〉 is a non-empty case-insensitive sequence of digits and letters
from A to F preceded by the (case sensitive) characters #x .

#x0 #xA04
#x01Ab #x61ff

Binaries. A 〈binary〉 is a non-empty sequence of the characters 0 and 1 preceded by the char-
acters #b .

#b0 #b1
#b001 #b101011

String literals. A 〈string〉 (literal) is any sequence of characters from 〈printable_char〉 or
〈white_space_char〉 delimited by the double quote character " (34dec). The character "
can itself occur within a string literal only if duplicated. In other words, after an initial "
that starts a literal, a lexer should treat the sequence "" as an escape sequence denoting
a single occurrence of " within the literal.

1Which implies that the language’s semantics does not depend on indentation and spacing.
2Note that the space character is both a printable and a whitespace character.

"this is a string literal"

""

"She said: ""Bye bye"" and left."

"this is a string literal

with a line break in it"

SMT-LIB string literals are akin to raw strings in certain programming languages. How-
ever, they have only one escape sequence: "" . This means, for example and in contrast
to most programming languages, that within a 〈string〉 the character sequences \n, \012,
\x0A, and \u0008 are not escape sequences (all denoting the new line character), but
regular sequences denoting their individual characters.(5)

Reserved words. The language uses a number of reserved words, sequences of printable char-
acters that are to be treated as individual tokens. The basic set of reserved words consists
of the following:

BINARY DECIMAL HEXADECIMAL NUMERAL STRING

_ ! as let exists forall match par

Additionally, each command name in the scripting language defined in Section 3.9
(set-logic, set-option, . . . ) is also a reserved word.(6)

Symbols. A 〈symbol〉 is either a simple symbol or a quoted symbol. A simple symbol is any
non-empty sequence of elements of 〈letter〉 and 〈digit〉 and the characters

~ ! @ $ % ^ & * _ - + = < > . ? /

that does not start with a digit and is not a reserved word.3

+ <= x plus ** $ <sas <adf >

abc77 *$s&6 .kkk .8 +34 -32

A quoted symbol is any sequence of whitespace characters and printable characters that
starts and ends with | and does not contain | or \ .(7)

3Note that simple symbols cannot contain non-English letters.

|this is a quoted symbol|

|so is
this one|

||

| " can occur too|

|af klj ^*0 asfe2 (&*)&(#^$>> >?" ’]]984|

Symbols are case sensitive. They are used mainly as operators or identifiers. Conven-
tionally, arithmetic characters and the like are used, individually or in combination, as
operator names; in contrast, alpha-numeric symbols, possibly with punctuation char-
acters and underscores, are used as identifiers. But, as in LISP, this usage is only rec-
ommended (for human readability), not prescribed. For additional flexibility, arbitrary
sequences of whitespace and printable characters (except for | and \) enclosed in verti-
cal bars are also allowed as symbols. Following Common Lisp’s conventions, enclosing
a simple symbol in vertical bars does not produce a new symbol. This means for instance that
abc and |abc| are the same symbol.

Simple symbols starting with the character @ or . are reserved for solver use.4 Solvers
can use them respectively as identifiers for abstract values and solver-generated function
symbols other than abstract values.

Keywords. A 〈keyword〉 is a token of the form :〈simple_symbol〉 . Elements of this category
have a special use in the language. They are used as attribute names or option names (see
later).

:date :a2 :foo -bar
:<= :56 :->

3.2 S-expressions

An S-expression is either a non-parenthesis token or a (possibly empty) sequence of S-ex-
pressions enclosed in parentheses. Every syntactic category of the SMT-LIB language is a
specialization of the category 〈s_expr〉 defined by the production rules below.

〈spec_constant〉 ::= 〈numeral〉 | 〈decimal〉 | 〈hexadecimal〉 | 〈binary〉 | 〈string〉
〈s_expr〉 ::= 〈spec_constant〉 | 〈symbol〉 | 〈keyword〉 | ( 〈s_expr〉∗ )

4This includes symbols such as |@abc| and |.abc| which are considered the same as @abc and .abc, respec-
tively.

Remark 1 (Meaning of special constants). Elements of the 〈spec_constant〉 category do not
always have the expected associated semantics in the SMT-LIB language (i.e., elements of
〈numeral〉 denoting integers, elements of 〈string〉 denoting character strings, and so on). In
particular, in the 〈term〉 category (defined later) they simply denote constant symbols, with
no fixed, predefined semantics. Their semantics is determined locally by each SMT-LIB theory
that uses them. For instance, it is possible for an SMT-LIB theory of sets to use the numerals 0
and 1 to denote respectively the empty set and universal set. Similarly, the elements of 〈binary〉
may denote integers modulo n in one theory and binary strings in another; the elements of
〈decimal〉 may denote rational numbers in one theory and floating point values in another.

3.3 Identifiers

Identifiers are used mostly as function and sort symbols. When defining certain SMT-LIB
theories it is convenient to have indexed identifiers as well. Instead of having a special token
syntax for that, indexed identifiers are defined more systematically as the application of the
reserved word _ to a symbol and one or more indices. Indices can be numerals or symbols.(8)

〈index〉 ::= 〈numeral〉 | 〈symbol〉
〈identifier〉 ::= 〈symbol〉 | ( _ 〈symbol〉 〈index〉+ )

plus + <= Real |John Brown|
(_ vector -add 4 5) (_ BitVec 32)
(_ move up) (_ move down) (_ move left) (_ move right)

We refer to identifiers from 〈symbol〉 as simple identifiers and to the others as indexed
identifiers. Since identifiers are used as the names of function symbols, sort symbols, sort
parameters, variables and commands, we often refer to them informally as names in this
document.

Remark 2 (Namespaces and shadowing of identifiers). There are several namespaces for iden-
tifiers: sorts, terms, command names, and attributes. The same identifier can occur in differ-
ent namespaces with no risk of conflicts because each namespace can always be identified
syntactically. Within the term namespace, bound variables can shadow one another as well
as function symbols in accordance with a lexical scoping discipline described in Section 3.6.
Similarly, sort parameters can shadow sort symbols, as described in Section 3.5.

3.4 Attributes

Several syntactic categories in the language contain attributes. These are generally pairs con-
sisting of an attribute name and an associated value, although attributes with no value are
also allowed.

Attribute names belong to the 〈keyword〉 category. Attribute values are in general S-
expressions other than keywords, although most predefined attributes use a more restricted
category for their values.

〈attribute_value〉 ::= 〈spec_constant〉 | 〈symbol〉 | ( 〈s_expr〉∗ )
〈attribute〉 ::= 〈keyword〉 | 〈keyword〉 〈attribute_value〉

:left -assoc
:status unsat
:my_attribute (humpty dumpty)
:authors "Jack and Jill"

3.5 Sorts

A major subset of the SMT-LIB language is the language of well-sorted terms, used to represent
logical expressions. Such terms are typed, or sorted in logical terminology; that is, each is
associated with a (unique) sort. The set of sorts consists itself of sort terms. In essence, a sort
term is a sort symbol, a sort parameter, or a sort symbol applied to a sequence of sort terms.

Syntactically, a sort symbol can be either the distinguished symbol Bool or any 〈identifier〉.
A sort parameter can be any 〈symbol〉 (which in turn, is an 〈identifier〉).

〈sort〉 ::= 〈identifier〉 | ( 〈identifier〉 〈sort〉+ )

Int Bool

(_ BitVec 3) (List (Array Int Real))

((_ FixedSizeList 4) Real) (Set (_ Bitvec 3))

A (non-)parametric sort is a sort containing (no) parameters. We will use σ to denote non-
parametric sorts and τ to denote possibly parametric sorts.

3.5.1 Ranks

Each function symbol in an SMT-LIB script is associated with one or more ranks, non-empty
sequences of sorts. Intuitively, a function symbol f with rank σ1 · · · σnσ, with none of the sorts
parametric, denotes a function that takes as input n values of respective sorts σ1, . . . , σn, and
returns a value of sort σ.

In contrast, a function symbol f with rank τ1 · · · τnτ, where each sort may contain sort
parameters, actually stands for a whole class of function symbols, all named f and each with a
rank obtained from τ1 · · · τnτ by instantiating in all possible ways every occurrence in τ1 · · · τnτ

of a sort parameter with a non-parametric sort.

3.6 Terms and Formulas

Abstractly, term are constructed out of constant symbols in the 〈spec_constant〉 category (nu-
merals, decimals, strings, etc.), variables, function symbols, four kinds of binders (introduced by

the reserved words let, forall, exists, and match), and an annotation operator (the reserved
word !). In its simplest form, a term is a special constant symbol, a variable, a function
symbol, or the application of a function symbol to one or more terms. More complex terms
include one or more binders.

Concretely, a variable can be any 〈symbol〉, while a function symbol can be any 〈identifier〉
(i.e., a symbol or an indexed symbol). As a consequence, contextual information is needed
during parsing to know whether an identifier is to be treated as a variable or a function
symbol. For variables, this information is provided by the three binders which are the only
mechanism to introduce variables. Function symbols, in contrast, are predefined, as explained
later. Recall that every function symbol f is separately associated with one or more ranks, each
specifying the sorts of f ’s arguments and result. To simplify sort checking, a function symbol
in a term can be annotated with one of its result sorts σ. Such an annotated function symbol
is a qualified identifier of the form (as f σ).

〈qual_identifier〉 ::= 〈identifier〉 | ( as 〈identifier〉 〈sort〉 )
〈var_binding〉 ::= ( 〈symbol〉 〈term〉 )
〈sorted_var〉 ::= ( 〈symbol〉 〈sort〉 )
〈pattern〉 ::= 〈symbol〉 | ( 〈symbol〉 〈symbol〉+ )

〈match_case〉 ::= ( 〈pattern〉 〈term〉 )
〈term〉 ::= 〈spec_constant〉

| 〈qual_identifier〉
| ( 〈qual_identifier〉 〈term〉+ )
| ( let ( 〈var_binding〉+ ) 〈term〉 )
| ( forall ( 〈sorted_var〉+ ) 〈term〉 )
| ( exists ( 〈sorted_var〉+ ) 〈term〉 )
| ( match 〈term〉 ( 〈match_case〉+ ) )
| ( ! 〈term〉 〈attribute〉+ )

SMT-LIB scripts can contain only well-sorted terms (see Section 3.6.4). Formulas in SMT-
LIB are just well-sorted terms of sort Bool. As a consequence, there is no syntactic distinction
between function and predicate symbols; the latter are simply function symbols whose result
sort is Bool. Another consequence is that function symbols can take formulas (even quantified
ones) as arguments.

(forall ((x (List Int)) (y (List Int)))
(= (append x y)

(ite (= x (as nil (List Int)))
y
(let ((h (head x)) (t (tail x)))

(insert h (append t y))))))

3.6.1 Variable Binders

Variables are introduced by means of one of the four binders. Each binder allows the intro-
duction of one or more variables or parameters with local scope.

Exists and forall quantifiers. These binders correspond to the usual universal and existential
quantifiers of first-order logic, except that each variable they quantify is also associated with
a sort. Both binders have a non-empty list of variables, which abbreviates a sequential nesting
of quantifiers. Specifically, a formula of the form

(forall ((x1 σ1) (x2 σ2) · · · (xn σn)) ϕ) (3.1)

has the same semantics as the formula

(forall ((x1 σ1)) (forall ((x2 σ2)) (· · · (forall ((xn σn)) ϕ) · · · ) (3.2)

Note that the variables in the list ((x1 σ1) (x2 σ2) · · · (xn σn)) of (3.1) are not required
to be pairwise disjoint. However, because of the nested quantifier semantics, earlier occur-
rences of same variable in the list are shadowed by the last occurrence—making those earlier
occurrences useless. The same argument applies to the exists binder.

Let. The let binder introduces and defines one or more local variables in parallel. Semanti-
cally, a term of the form

(let ((x1 t1) · · · (xn tn)) t) (3.3)

is equivalent to the term t[t1/x1, . . . , tn/xn] obtained from t by simultaneously replacing each
free occurrence of xi in t by ti, for each i = 1, . . . , n, possibly after a suitable renaming of
t’s bound variables to avoid capturing any variables in t1, . . . , tn. Because of the parallel
semantics, the variables x1, . . . , xn in (3.3) must be pairwise distinct.

Remark 3 (No sequential version of let). The language does not have a sequential version of
let. Its effect is achieved by nesting lets, as in (let ((x1 t1)) (let ((x2 t2)) t)).

Match. Similarly to pattern matching statements in functional programming languages or
in certain interactive theorem provers, the match binder is used to perform pattern matching
on values of an algebraic data type (see Section 4.2.3). It has the form

(match t ((p1 t1) · · · (pm+1 tm+1))) (3.4)

where t is a term of some datatype sort δ and, for each i = 1, . . . , m + 1, pi is a pattern for δ,
and ti a term of some sort σ. A pattern p in turn is either a variable x of sort δ, a nullary
constructor c of δ, or a term of the form (c x1 · · · xk) where c is a constructor of δ of rank
σ1 · · · σkδ with k > 0, and x1, . . . , xk are distinct variables of respective sort σ1, . . . , σk.(9) The list
p1, . . . pm+1 may contain more than one pattern with the same constructor or more than one
pattern consisting of a variable.(10) However, it must contain a pattern consisting of a variable,
unless every constructor of δ occurs in one of the patterns.(11)

The term ti in (3.4) can contain free occurrences of the variables occurring in pattern pi, if
any. The scope of those variables is the term ti.

; Axiom for list append: version 1
; List is a parametric datatype
; with constructors "nil" and "cons"
;
(forall ((l1 (List Int)) (l2 (List Int )))

(= (append l1 l2)
(match l1 (

(nil l2)
((cons h t) (cons h (append t l2 )))))))

; Axiom for list append: version 2
(forall ((l1 (List Int)) (l2 (List Int )))

(= (append l1 l2)
(match l1 (

((cons h t) (cons h (append t l2)))
(_ l2 ))))) ; _ is a variable

Remark 4 (No nested patterns). Nested patterns, where the arguments of a constructor can
be themselves non-variable patterns, are not allowed.(12) This implies in particular that in a
pattern of the form (c s1 · · · sk), where s1, . . . , sk are symbols, those symbols are always to
be parsed as variables.

Strictly speaking, the match binder is not essential since it can be defined in terms of the
other binders. Specifically, expression (3.4) can be written equivalently as follows.

1. Suppose that for all i = 1, . . . , m + 1, the pattern pi has the form (ci xi,1 · · · xi,ki) with
ki > 0; qi is the associate tester for that constructor; and si,1, . . . , si,ki are the selectors
associated, in order, with the constructor’s arguments. Then, match expression (3.4) has
the same meaning as

(ite (q1 t) (let ((x1,1 (s1,1 t)) · · · (x1,k1 (s1,k1 t))) t1)
(ite (q2 t) (let ((x2,1 (s2,1 t)) · · · (x2,k2 (s2,k2 t))) t2)
· · ·
(ite (qm t) (let ((xm,1 (sm,1 t)) · · · (xm,km (sm,km t))) tm)
(let ((xm+1,1 (sm+1,1 t)) · · · (xm+1,km+1 (sm+1,km+1 t))) tm+1) · · · )

(3.5)

2. When for some i ∈ {1, . . . , m+ 1}, the pattern pi is a nullary constructor, the correspond-
ing let subexpression in (3.5) is replaced by ti.

3. If instead for some minimal i ∈ {1, . . . , m + 1}, the pattern pi is a variable x, then the
whole ith ite subexpression of (3.5), if i ≤ m, or the let subexpression, if i = m, is
replaced by (let ((x t)) ti).

3.6.2 Parameter Binders

SMT-LIB allows the declaration of parametric function symbols in theory declarations (see
Section 3.7) and of parametric datatypes in user scripts (see Section 4.2.3). Sort parameters are
introduced with expressions of the form

(par (u1 · · · uk) e)

where u1, . . . , uk are k > 0 pairwise distinct sort parameters and e is an expression containing
occurrences of u1, . . . , uk.

3.6.3 Scoping of variables and parameters

The notions of free variable/parameter occurrence in an s-expression, bound variable/param-
eter, and (binder) scope are defined as follows.

A variable x:

• occurs free in the expression x;

• occurs free in an expression (e1 · · · en) if e1 is par or not a binder, and x occurs free in
some ei (1 ≤ i ≤ n);

• occurs free in an expression (forall (s1 · · · sn) t) or (exists (s1 · · · sn) t) if it
does not occur in s1, . . . , sn and occurs free in t;

• occurs free in an expression (let ((x1 t1) · · · (xn tn)) t) if (i) it occurs free in some
ti (1 ≤ i ≤ n) and the corresponding xi occurs free in t, or (ii) it does not occur in
x1, . . . , xn and occurs free in t;

• occurs free in an expression (match t ((p1 t1) · · · (pn tn))) if it occurs free in t or it
occurs free in some ti (1 ≤ i ≤ n) and does not occur in the corresponding pi.

A parameter u:

• occurs free in the expression u;

• occurs free in an expression (e1 · · · en) if e1 is not par and u occurs free in some ei
(1 ≤ i ≤ n);

• occurs free in an expression (par (u1 · · · un) e) if it occurs free in e and does not occur
in (u1 · · · un).

Each non-free, or bound, occurrence of a variable in an expression has a scope defined as
follows.

• In an expression (q ((x1 σ1) (x2 σ2) · · · (xn σn)) t) where q is either forall or
exists, or in an expression (let ((x1 t1) · · · (xn tn)) t), the scope of each variable
in {x1, . . . , xn} is the term t.

• In an expression (match t ((p1 t1) · · · (pn tn))), the scope of each variable occurring
in pattern pi is the corresponding term ti (1 ≤ i ≤ n).

• In an expression (par (u1 · · · un) e), the scope of each parameter in {u1, . . . , un} is
the expression e.

Shadowing All binders follow a lexical scoping discipline, consistently with the semantics of
the SMT-LIB logic, as described in Section 5.3. In particular, a bound variable or parameter will
shadow any variable or user-defined function symbol with the same name from an enclosing
scope. For instance, in a function symbol declaration like (par (Int) (f (Array Int Int)
Int)), all occurrences of Int are sort parameters regardless of the existence of a previously
defined sort symbol Int.(13) Similarly, in a match pattern like (cons x nil), the symbol nil
is a variable regardless of the existence of a previously defined constructor symbol nil.5

Remark 5 (No shadowing of theory symbols). One exception is that binders cannot shadow
theory function symbols, that is, function symbols from the declaration (see Section 3.7) of a
theory included in the current logic (see Section 3.8 and Subsection 4.2.1). In other words,
variables and parameters cannot have the same name as a theory function symbol in the same
scope.(14)

3.6.4 Well-sortedness requirements

All terms of the SMT-LIB language are additionally required to be well sorted. Well-sortedness
rules are presented and discussed in Section 5.2, in terms of the logic’s abstract syntax.

Except for patterns in match expressions, every occurrence of an ambiguous function sym-
bol6 f in a term must occur as a qualified identifier of the form (as f σ) where σ is the
intended output sort of that occurrence.(15) The same requirement applies to occurrences of
solver-generated constants within terms output by the solver—as the type of these constants
is unknown to the user.

(cons "abc" (as nil (List String )))

(= a (as (const -array 0.0) (Array Int Real )))

(select (as @a1 (Array Int Int)) 3)

3.6.5 Annotations

Every term t can be optionally annotated with one or more attributes α1, . . . , αn using the
wrapper expression (! t α1 · · · αn). Term attributes have no logical meaning—semantically,
(! t α1 · · · αn) is equivalent to t—but they are a convenient mechanism for adding meta-
logical information for SMT solvers.

3.6.6 Term attributes

Currently there are only two predefined term attributes: :named and :pattern. The values
of the :named attribute range over the 〈symbol〉 category. The attribute can be used in scripts

5Note that a pattern (cons x nil) where nil was a constructor would be nested, which is not allowed.
6That is, function symbols with more than one possible output sort for the same sequence of input sorts (see

Subsection 5.2.1).

to give a closed term a symbolic name, which can then be used as a proxy for the term (see
Section 4.2).

(=> (! (> x y) :named p1)
(! (= x z) :named p2))

The values of the :pattern attribute range over sequences of 〈term〉 elements. The attribute
is used to define instantiation patterns for quantifiers, which provide heuristic information to
SMT solvers that reason about quantified formulas by quantifier instantiation. Instantiation
patterns can only be used to annotate the body ϕ of a quantified formula of the form

(Q ((x1 σ1) · · · (xk σk)) ϕ)

where Q is forall or exists, so that the resulting annotated formula has the form

(Q ((x1 σ1) · · · (xk σk)) (! ϕ :pattern (p1,1 · · · p1,n1)
...

:pattern (pm,1 · · · pm,nm)))

where each pi,j is a binder-free term with no annotations and the same well-sortedness re-
quirements as the formula’s body.(16)

(forall ((x0 A) (x1 A) (x2 A))
(! (=> (and (r x0 x1) (r x1 x2)) (r x0 x2))
:pattern ((r x0 x1) (r x1 x2))
:pattern ((p x0 a))

))

The intended use of these patterns is to suggest to the solver that it should try to find inde-
pendently for each i = 1, . . . , m, a sequence ti,1 · · · ti,ni of ni ground terms that simultaneously
match the pattern terms pi,1 · · · pi,ni .

7

3.7 Theory Declarations

The set of SMT-LIB theories is defined by a catalog of theory declarations written in the format
specified in this section. This catalog is available on the SMT-LIB web site at www.smt-
lib.org. In earlier versions of the SMT-LIB standard, a theory declaration defined both a many-
sorted signature, i.e., a collection of sorts and sorted function symbols, and a theory with that
signature. The signature was determined by the collection of individual declarations of sort
symbols and function symbols with an associated rank—specifying the sorts of the symbol’s
arguments and of its result.

From Version 2.0 on, theory declarations may also declare entire families of overloaded
function symbols by using ranks that contain sort parameters, locally scoped sort symbols of

7The terms assigned to the variables x1, . . . , xk by the simultaneous matching substitution are typically used to
instantiate the body of a universally quantified formula in order to generate ground consequences of that formula.

http://www.smt-lib.org
http://www.smt-lib.org

〈sort_symbol_decl〉 ::= ( 〈identifier〉 〈numeral〉 〈attribute〉∗ )
〈meta_spec_constant〉 ::= NUMERAL | DECIMAL | STRING

〈fun_symbol_decl〉 ::= ( 〈spec_constant〉 〈sort〉 〈attribute〉∗ )
| ( 〈meta_spec_constant〉 〈sort〉 〈attribute〉∗ )
| ( 〈identifier〉 〈sort〉+ 〈attribute〉∗ )

〈par_fun_symbol_decl〉 ::= 〈fun_symbol_decl〉
| ( par ( 〈symbol〉+ ) ( 〈identifier〉 〈sort〉+ 〈attribute〉∗ ) )

〈theory_attribute〉 ::= :sorts ( 〈sort_symbol_decl〉+ )
| :funs ( 〈par_fun_symbol_decl〉+ )
| :sorts-description 〈string〉
| :funs-description 〈string〉
| :definition 〈string〉
| :values 〈string〉
| :notes 〈string〉
| 〈attribute〉

〈theory_decl〉 ::= ( theory 〈symbol〉 〈theory_attribute〉+ )

Figure 3.1: Theory declarations.

arity 0. This kind of parametricity entails that a theory declaration generally defines a whole
class of similar theories.

The syntax of theory declarations, specified in Figure 3.1, follows an attribute-value-based
format. A theory declaration consists of a theory name and a list of 〈attribute〉 elements.
Theory attributes with the following keywords are predefined attributes, with prescribed usage
and semantics:

:definition :funs :funs-description
:notes :sorts :sorts-description :values .

Additionally, a theory declaration can contain any number of user-defined attributes.(17)

Theory attributes can be formal or informal depending on whether or not their values
have a formal semantics and can be processed in principle automatically. The value of an
informal attribute is free text, in the form of a 〈string〉 literal or a quoted symbol. For in-
stance, the attributes :funs and :sorts are formal in the sense above, whereas :definition,
:funs-description and :sorts-description are not.

A theory declaration (theory T α1 · · · αn) defines a theory schema with name T and at-
tributes α1, . . . , αn. Each instance of the schema is a theory TΣ with an expanded signature
Σ, containing (zero or more) additional sort and function symbols with respect to those de-
clared in T. Theories are defined as classes of first-order structures (or models) of signature Σ.
See Section 5.4 for a formal definition of theories and a more detailed explanation of how a
theory declaration can be instantiated to a theory. Concrete examples of instances of theory
declarations are discussed later.

The value of a :sorts attribute is a non-empty sequence of sort symbol declarations
〈sort_symbol_decl〉. A sort symbol declaration (s n α1 · · · αm) declares a sort symbol s

of arity n, and may additionally contain zero or more annotations, each in the form of an
〈attribute〉. In this version, there are no predefined annotations for sort declarations.

The value of a :funs attribute is a non-empty sequence of possibly parametric function
symbol declarations 〈par_fun_symbol_decl〉. A (non-parametric) function symbol declaration
〈fun_symbol_decl〉 of the form (c σ), where c is an element of 〈spec_constant〉, declares c to
have sort σ. For convenience, it is possible to declare all the special constants in 〈numeral〉
to have sort σ by means of the function symbol declaration (NUMERAL σ). This is done for
instance in the theory declaration in Figure 3.3. The same can be done for the set of 〈decimal〉
and 〈string〉 constants by using DECIMAL and STRING, respectively.

A (non-parametric) function symbol declaration of the form

( f σ1 · · · σn σ)

with n ≥ 0 declares a function symbol f with rank σ1 · · · σnσ.
A parametric function symbol declaration, which has the form,

(par (u1 · · · uk) ( f τ1 · · · τn τ))

with k > 0, n ≥ 0 and u1, . . . , uk all distinct, declares a function symbol f with parametric
rank τ1 · · · τnτ. This effectively declares a whole class of function symbols, all named f and
each with a rank obtained from τ1 · · · τnτ by instantiating in all possible ways each occurrence
in τ1 · · · τnτ of the sort parameters u1, . . . , uk with non-parametric sorts.

As with sorts, each (parametric) function symbol declaration may additionally contain
zero or more annotations α1, . . . , αn, each in the form of an 〈attribute〉. In this version, there
are only 4 predefined function symbol annotations, all attributes with no value: :chainable,
:left-assoc, :right-assoc, and :pairwise. The :left-assoc annotation can be added only
to function symbol declarations of the form

( f σ1 σ2 σ1) or (par (u1 · · · uk) ( f τ1 τ2 τ1)).

Then, an expression of the form ( f t1 · · · tn) with n > 2 is allowed as syntactic sugar (re-
cursively) for ( f ( f t1 · · · tn−1) tn). Similarly, the :right-assoc annotation can be added
only to function symbol declarations of the form

( f σ1 σ2 σ2) or (par (u1 · · · uk) ( f τ1 τ2 τ2)).

Then, ( f t1 · · · tn) with n > 2 is syntactic sugar for ( f t1 ( f t2 · · · tn)).
The :chainable and :pairwise annotations can be added only to function symbol decla-

rations of the form

( f σ σ Bool) or (par (u1 · · · uk) ( f τ τ Bool))

and are mutually exclusive. With the first annotation, ( f t1 · · · tn) with n > 2 is syntactic
sugar for (and ( f t1 t2) ( f t2 t3) · · · ( f tn−1 tn)) where and is itself a symbol declared as
:left-assoc in every theory (see Subsection 3.7.1); with the second, ( f t1 · · · tn) is syntactic
sugar (recursively) for (and ( f t1 t2) · · · ( f t1 tn) ( f t2 · · · tn)).

(+ Real Real Real :left -assoc)

(and Bool Bool Bool :left -assoc)

(par (X) (insert X (List X) (List X) :right -assoc))

(< Real Real Bool :chainable)

(equiv Elem Elem Bool :chainable)

(par (X) (Disjoint (Set X) (Set X) Bool :pairwise ))

(par (X) (distinct X X Bool :pairwise ))

For many theories in SMT-LIB, in particular those with a finite signature, it is possible to
declare all of their symbols using a finite number of sort and function symbol declarations
in :sorts and :funs attributes. For others, such as, for instance, the theory of bit vectors,
one would need infinitely many such declarations. In those cases, sort symbols and function
symbols are defined informally, in plain text, in :sorts-description, and :funs-description
attributes, respectively.(18)

:sorts_description
"All sort symbols of the form (_ BitVec m) with m > 0."

:funs_description
"All function symbols with rank of the form

(concat (_ BitVec i) (_ BitVec j) (_ BitVec m))

where i,j > 0 and i + j = m."

The :definition attribute is meant to contain a natural language definition of the theory.
While this definition is expected to be as rigorous as possible, it does not have to be a formal
one.(19) For some theories, a mix of formal notation and natural language might be more
appropriate. In the presence of parametric function symbol declarations, the definition must
also specify the meaning of each instance of the declared symbol.(20)

The attribute :values is used to specify, for each sort σ, a distinguished, decidable set of
ground terms of sort σ that are to be considered as values for σ. We will call these terms value
terms. Intuitively, given an instance theory containing a sort σ, σ’s set of value terms is a set
of terms that denotes, in each countable model of the theory, all the elements of that sort.
These terms might be over a signature with additional function symbols with respect to those
specified in the theory declaration. Ideally, the set of value terms is minimal, which means
that no two distinct terms in the set denote the same element in some model of the theory.
However, this is only a recommendation, not a requirement because it is impractical, or even

impossible, to satisfy it for some theories. See the next subsection for examples of value sets,
and Section 5.5 for a more in-depth explanation.

The attribute :notes is meant to contain documentation information on the theory decla-
ration such as authors, date, version, references, etc., although this information can also be
provided with more specific, user-defined attributes.

Constraint 1 (Theory Declarations). The only legal theory declarations of the SMT-LIB lan-
guage are those that satisfy the following restrictions.

1. They contain exactly one occurrence of the :definition and the :values attribute8 and
any number of occurrences of other attributes.

2. Each sort symbol used in a :funs attribute is previously declared in some :sorts at-
tribute. In each parametric function symbol declaration (par (u1 · · · uk) ( f τ1 · · · τn

τ)), any symbol other than f that is not a previously declared sort symbol must be one
of the sort parameters u1, . . . , uk.

3. The definition of the theory, provided in the :definition attribute, refers only to sort
and function symbols previously declared formally in :sorts and :funs attributes or
informally in :sorts-description and :funs-description attributes.

Note that the :funs attribute is optional in a theory declaration because a theory might
lack function symbols (although such a theory would not be not very interesting).

3.7.1 Examples

Core theory

To provide the usual set of Boolean connectives for building formulas, in addition to the
predefined logical symbol distinct, a basic core theory is implicitly included in every other
SMT-LIB theory. Concretely, every theory declaration is assumed to contain implicitly the
:sorts and :funs attributes of the Core theory declaration shown in Figure 3.2, and to define
the symbols in those attributes in the same way as in Core.

Note the absence of a symbol for double implication. Such a connective is superfluous
because the equality symbol = can be used in its place. Note how the attributes specified
in the declarations of the various symbols of this theory allow one, for instance, to write
expressions like

(=> x y z)

(and x y z)

(= x y z)

(distinct x y z)

respectively as abbreviations for the terms

8Which makes those attributes non-optional.

(theory Core

:sorts ( (Bool 0) )

:funs ( (true Bool) (false Bool) (not Bool Bool)
(=> Bool Bool Bool :right-assoc) (and Bool Bool Bool :left-assoc)
(or Bool Bool Bool :left-assoc) (xor Bool Bool Bool :left-assoc)
(par (A) (= A A Bool :chainable))
(par (A) (distinct A A Bool :pairwise))
(par (A) (ite Bool A A A)) )

:definition
"For every expanded signature Sigma, the instance of Core with that signature
is the theory consisting of all Sigma-models in which:
- the sort Bool denotes the set {true, false} of Boolean values;
- for all sorts s in Sigma,
- (= s s Bool) denotes the function that
returns true iff its two arguments are identical;

- (distinct s s Bool) denotes the function that
returns true iff its two arguments are not identical;

- (ite Bool s s) denotes the function that
returns its second argument or its third depending on whether
its first argument is true or not;

- the other function symbols of Core denote the standard Boolean operators
as expected."

:values "The set of values for the sort Bool is {true, false}."
)

Figure 3.2: The Core theory declaration.

(=> x (=> y z))

(and (and x y) z)

(and (= x y) (= y z))

(and (distinct x y) (distinct x z) (distinct y z)) .

The simplest instance of Core is the theory with no additional sort and function symbols.
In that theory, there is only one sort, Bool, and ite has only one rank, (ite Bool Bool Bool
Bool). In other words, this is just the theory of the Booleans with the standard Boolean oper-
ators plus ite and distinct. The set of values for the Bool sort is, predictably, {true, false}.

Another instance has a single additional sort symbol U, say, of arity 0, and a (possibly
infinite) set number of function symbols with rank in U +. This theory corresponds to EUF,
the (one-sorted) theory of equality and uninterpreted functions (over those function symbols).
In this theory, ite has two ranks: (ite Bool Bool Bool Bool) and (ite Bool U U U). A
many-sorted version of EUF is obtained by instantiating Core with more than one nullary sort
symbol—and possibly additional function symbols over the resulting sort set.

(theory Ints
:sorts ( (Int 0) )
:funs ( (NUMERAL Int)

(- Int Int) ; negation
(- Int Int Int :left-assoc) ; subtraction
(+ Int Int Int :left-assoc)
(* Int Int Int :left-assoc)
(<= Int Int Bool :chainable)
(< Int Int Bool :chainable)
(>= Int Int Bool :chainable)
(> Int Int Bool :chainable) )

:definition
"For every expanded signature Sigma, the instance of Ints with that
signature is the theory consisting of all Sigma-models that interpret
- the sort Int as the set of all integers,
- the function symbols of Ints as expected. "

:values
"The Int values are all the numerals and all the terms of the form (- n)
where n is a non-zero numeral."

)

Figure 3.3: A possible theory declaration for the integer numbers.

Yet another instance is the theory with an additional unary sort symbol List and an addi-
tional number of function symbols. This theory has infinitely many sorts: Bool, (List Bool),
(List (List Bool)), etc. However, by the definition of Core, all those sorts and function
symbols are still “uninterpreted” in the theory. In essence, this theory is the same as a many-
sorted version of EUF with infinitely many sorts. While not very interesting in isolation, the
theory is useful in combination with a theory of lists that, for each sort σ, interprets (List
σ) as the set of all lists over σ. The combined theory in that case is a theory of lists with
uninterpreted functions.

Integers

The theory declaration of Figure 3.3 defines all theories that extend the standard theory of
the (mathematical) integers with additional uninterpreted sort and function symbols.9 The
integers theory proper is the instance with no additional symbols. More precisely, since the
Core theory declaration is implicitly included in every theory declaration, that instance is the
two-sorted theory of the integers and the Booleans. The set of values for the Int sort consists
of all numerals and all terms of the form (- n) where n is a numeral other than 0.

(theory ArraysEx
:sorts ( (Array 2) )
:funs ( (par (X Y) (select (Array X Y) X Y))

(par (X Y) (store (Array X Y) X Y (Array X Y))) )
:notes
"A schematic version of the theory of functional arrays with extensionality."

:definition
"For every expanded signature Sigma, the instance of ArraysEx with that
signature is the theory consisting of all Sigma-models that satisfy all
axioms of the form below, for all sorts s1, s2 in Sigma:

- (forall ((a (Array s1 s2)) (i s1) (e s2))
(= (select (store a i e) i) e))

- (forall ((a (Array s1 s2)) (i s1) (j s1) (e s2))
(=> (distinct i j) (= (select (store a i e) j) (select a j))))

- (forall ((a (Array s1 s2)) (b (Array s1 s2)))
(=>
(forall ((i s1)) (= (select a i) (select b i))) (= a b))) "

:values
"For all sorts s1, s2, the values of sort (Array s1 s2) are either abstract
or have the form (store a i v) where
- a is value of sort (Array s1 s2),
- i is a value of sort s1, and
- v is a value of sort s2."

)

Figure 3.4: The ArraysEx theory declaration.

Arrays with extensionality

A schematic version of the theory of functional arrays with extensionality is defined in the
theory declaration ArraysEx in Figure 3.4. Each instance gives a theory of (arbitrarily nested)
arrays. For instance, with the addition of the nullary sort symbols Int and Real, we get an
instance theory whose sort set S contains, inductively, Bool, Int, Real and all sorts of the form
(Array σ1 σ2) with σ1, σ2 ∈ S. This includes flat array sorts such as

(Array Int Int), (Array Int Real), (Array Real Int), (Array Bool Int),

conventional nested array sorts such as

(Array Int (Array Int Real)),

as well as nested sorts such as

(Array (Array Int Real) Int), (Array (Array Int Real) (Array Real Int))

with an array sort in the index position of the outer array sort.(21)

The function symbols of the theory include all symbols with name select and rank of the
form ((Array σ1 σ2) σ1 σ2) for all σ1, σ2 ∈ S. Similarly for store.

9For simplicity, the theory declaration in the figure is an abridged version of the declaration actually used in
the SMT-LIB catalog.

Sets and Relations

A schematic many-sorted version of the theory of hereditary well-founded sets with urele-
ments is defined in the theory declaration SetsRelations in Figure 3.5. Each instance gives a
theory of sets of elements of the same sort. These elements can be either atomic (i.e., of a prim-
itive sort like Bool), or tuples of elements, or sets themselves. For instance, with the addition
of the nullary sort symbol Int we get an instance theory whose sort set S contains, induc-
tively, Bool, Int and all sorts of the form (Set σ) or (Prod σ1 · · · σn) with σ, σ1, . . . , σn ∈ S.
In each model of the theory that interprets Int as the integers, we get Booleans, integers, sets
of Booleans, sets of integers, sets of tuples over these sets, and so on.

Note that every sort σ in the signature is internalized in the theory, since the set denoted
by σ is also denoted by the constant univSet of (the powerset) sort (Set σ).

Remark 6 (Instances of parametric sorts). For some applications, the instantiation mechanism
defined here for theory declarations will definitely over-generate. For instance, it is not pos-
sible to define by instantiation of the ArraysEx declaration a theory of just the arrays of sort
(Array Int Real), without all the other nested array sorts over {Int, Real}. This, however,
is a problem neither in theory nor in practice. It is not a problem in practice because, since
a script can only use formulas with non-parametric sorts,10 any sorts that are not used in a
script are, for practical purposes, irrelevant. It is not a problem in theory either because scripts
refer to logics, not directly to theories. And the language of a logic can be always restricted to
contain only a selected subset of the sorts in the logic’s theory.

3.8 Logic Declarations

The SMT-LIB format allows the explicit definition of sublogics of its main logic— a version
of many-sorted first-order logic with equality—that restrict both the main logic’s syntax and
semantics. A new sublogic, or simply logic, is defined in the SMT-LIB language by a logic
declaration; see www.smt-lib.org for the current catalog. Logic declarations have a similar
format to theory declarations, although most of their attributes are informal.(22)

Attributes with the following predefined keywords are predefined attributes, with prescribed
usage and semantics in logic declarations:

:theories :language :extensions :notes :values .

Additionally, as with theories, a logic declaration can contain any number of user-defined
attributes.

〈logic_attribute〉 := :theories ( 〈symbol〉+ )
| :language 〈string〉
| :extensions 〈string〉
| :values 〈string〉
| :notes 〈string〉
| 〈attribute〉

〈logic〉 ::= ( logic 〈symbol〉 〈logic_attribute〉+ )

10Note that sort parameters cannot occur in a formula.

http://www.smt-lib.org

(theory SetsRelations
:sorts ( (Set 1) )
:funs ( (par (X) (emptySet (Set X)))

(par (X) (univSet (Set X)))
(par (X) (singleton X (Set X)))
(par (X) (union (Set X) (Set X) (Set X) :left-assoc))
(par (X) (inters (Set X) (Set X) (Set X) :left-assoc))
(par (X) (in X (Set X) Bool))
(par (X) (subset (Set X) (Set X) Bool :chainable)) )

:sorts_description
"All sort symbol declarations of the form (Prod n) with n > 1"

:funs_description
"All function symbols with declarations of the form

(par (X1 ... Xn) (tuple X1 ... Xn (Prod X1 ... Xn)))
(par (X1 ... Xn) ((_ project i) (Prod X1 ... Xn) Xi))
(par (X1 ... Xn) (prod (Set X1) ... (Set Xn) (Set (Prod X1 ... Xn))))

with n > 1 and i = 1,...,n"
:notes
"A schematic theory of sets and relations."

:definition
"For every expanded signature Sigma, the instance of SetsRelations
with that signature is the theory consisting of all Sigma-models that
for all sorts s, s1,..., sn, with n > 1, interpret
- (Set s) as the powerset of the set denoted by s
- (as emptySet (Set s)) as the empty set of sort (Set s)
- (as univSet (Set s)) as the set denoted by s
- (Prod s1 ... sn) as the Cartesian product of the sets denoted by s1,...,sn
- (tuple s1 ... sn (Prod s1 ... sn)) as the function that maps
its inputs x1, ..., xn to the tuple (x1, ..., xn)

- ((_ project i) (Prod s1 ... sn) si) for i = 1, ..., n as the i-th
projection function

- (prod (Set s1) ... (Set sn) (Set (Prod s1 ... sn))) as the function
that maps its input sets to their Cartesian product

and interpret the other function symbols as the corresponding set operators
as expected."

)

Figure 3.5: A possible declaration for a theory of sets and relations.

A logic declaration (logic L α1 · · · αn) defines a logic with name L and attributes
α1, . . . , αn.

Constraint 2 (Logic Declarations). The only legal logic declarations in the SMT-LIB language
are those that satisfy the following restrictions:

1. They include exactly one occurrence of the :theories and the :language attribute (and
any number of occurrences of other attributes).

2. The value (T1 · · · Tn) of the :theories attribute lists names of theory schemas that
have a declaration in SMT-LIB.

3. If two theory declarations among T1, . . . , Tn declare the same sort symbol, they give it
the same arity.

When the value of the :theories attribute is (T1 · · · Tn), with n > 0, the logic refers
to a combination T of specific instances of the theory declaration schemas T1, . . . , Tn. The
exact combination mechanism that yields T is defined formally in Section 5.5. The effect
of this attribute is to declare that the logic’s sort and function symbols consist of those of the
combined theory T , and that the logic’s semantics is restricted to the models of T , as specified
in more detail in Section 5.5.

The :language attribute describes in free text the logic’s language, a specific class of SMT-
LIB formulas. This information is useful for tailoring SMT solvers to the specific sublanguage
of formulas used in an input script.(23) The formulas in the logic’s language are built over
(a subset of) the signature of the associated theory T , as specified in this attribute. In the
context of a command script the language of a logic is implicitly expanded by let constructs
in formulas as well as user-defined (but not user-declared) sort and function symbols. In other
words, a formula ϕ used in a script is considered to belong to a certain logic’s language iff
the formula obtained from ϕ by replacing all let variables and all defined sort and function
symbols by their respective definitions is in the language.

The optional :extensions attribute is meant to document any notational conventions or
syntactic sugar allowed in the concrete syntax of formulas in this logic.(24)

The :values attribute has the same use as in theory declarations but it refers to the specific
theories and sorts of the logic. It is meant to complement the :values attributes specified in
the theory declarations referred to in the :theories attribute.

The textual :notes attribute serves the same purpose as in theory declarations.

3.8.1 Examples

Defining theories model-theoretically, as opposed to axiomatically as in more traditional ap-
proaches, confers great expressive power to the SMT-LIB underlying logic in spite of its re-
striction to a first-order syntax. Several established logics, from propositional all the way to
higher-order logic, can be defined as SMT-LIB sublogics given a suitable theory. We provide
a small sample below, for illustrative purposes. Again, see www.smt-lib.org for the list of the
actual logics defined in the SMT-LIB standard.

http://www.smt-lib.org

Propositional logic

Propositional logic can be readily defined by an SMT-LIB logic declaration. The logic’s theory
is the instance of the Core theory declaration whose signature adds infinitely-many function
symbols of rank Bool (playing the role of propositional variables). The language consists of all
binder-free formulas over the expanded signature. Extending the language with let binders
allows a faithful encoding of binary decision diagrams (BDDs) as formulas, thanks to the ite
operator of Core.

Quantified Boolean logic

The logic of quantified Boolean formulas (QBFs) can be defined as well. The theory is again
an instance of Core but this time with no additional symbols at all. The language consists of
(closed) quantified formulas all of whose variables are of sort Bool.

Linear integer arithmetic

Linear integer arithmetic can be defined as an SMT-LIB logic. This logic is indeed part of the
official SMT-LIB catalog of logics and is called QF_LIA there. Its theory is an extension of the
theory of integers and the Booleans with uninterpreted constant symbols. That is, it is the
instance of the theory declaration Ints from Figure 3.3 whose signature adds to the symbols
of Ints infinitely many free constants, new function symbols of rank Int or of rank Bool.

The language of the logic is made of closed quantifier-free formulas (over the theory’s
signature) containing only linear atoms, that is, atomic formulas with no occurrences of the
function symbol *. Extensions of the basic language include expressions of the form (* n
t) and (* t n), for some numeral n, both of which abbreviate the term (+ t · · · t) with n
occurrences of t (or 0 if n is 0). Also included are terms with negative integer coefficients, that
is, expressions of the form (* (- n) t) or (* t (- n)) for some numeral n, both of which
abbreviate the expression (- (* n t)).

Higher-order logic

The full language of the theory SetsRelations from Figure 3.5 has the same expressive power
as higher-order logic or, more precisely, higher-order simple predicate logic with primitive
types.11 Higher-order quantification is realized by first-order quantification over sorts of the
form (Set σ). Similarly, higher-order predicate symbols can be encoded as unary first-order
predicate symbols over sets. In fact, because of the presence of the predicate symbol in, no
additional predicate symbols at all are needed: every atomic formula of the form (p t) can be
replaced by the formula (in t p′) where p′ is a constant symbol of sort (Set (Prod σ1 · · ·
σn)) if p has rank ((Prod σ1 · · · σn) Bool), and of sort (Set σ) if p has rank (σ Bool).

11This is a relational version of higher-order logic, as opposed to the functional versions that are more popular
in computer science.

Monadic second-order logic

This logic can be obtained by imposing restrictions on the language of the SetsRelations
theory mentioned above, specifically, by disallowing the use of terms of Prod sorts (monadic
restriction) or nested Set sorts (second-order restriction) while, however, allowing additional
function and predicate symbols.

3.9 Scripts

Scripts are sequences of commands. In line with the LISP-like syntax, all commands look like
LISP-function applications, with a command name applied to zero or more arguments. To
facilitate processing, each command takes a constant number of arguments, although some
of these arguments can be (parenthesis-delimited) lists of variable length. The full list of
commands is provided in Figure 3.6.

The intended use of scripts is to communicate with an SMT-solver in a read-eval-print loop:
until a termination condition occurs, the solver reads the next command, acts on it, outputs a
response, and repeats. Possible responses vary from a single symbol to a list of attributes, to
complex expressions like proofs.

The command set-option takes as an argument expressions of the syntactic category
〈option〉, which have the same form as attributes with values. Options with the predefined
keywords listed in Figure 3.7 have a prescribed usage and semantics. Additional, solver-
specific options are also allowed.

The command get-info takes as argument expressions of the syntactic category 〈info_flag〉
which are flags with the same form as keywords. The predefined flags listed in Figure 3.8 have
a prescribed usage and semantics. Additional, solver-specific flags are also allowed. Exam-
ples of the latter might be, for instance, flags such as :time and :memory, referring to used
resources, or :decisions, :conflicts, and :restarts, referring to typical statistics for current
SMT solvers.

For more on error behavior, the meanings of the various options and info names, and the
semantics of the various commands, see Chapter 4. We highlight a few salient points here and
provide a couple of examples.

Assertion stack. Compliant solvers respond to various commands mostly by performing op-
erations on a data structure we call the assertion stack. This is a single stack whose elements,
called levels, are sets of assertions. Assertions include logical formulas (that is, terms of sort
Bool), as well as declarations and definitions of sort and function symbols. Assertions are
added by specific commands. By default, an assertion belongs to the most recent level at the
time the corresponding command was executed. The stack starts with a first assertion level
that cannot be removed. Further levels can be introduced by a push command and removed
by a corresponding pop command. Popping a level from the assertion stack has the effect
of undoing all assertions in it, including symbol declarations and definitions. An input op-
tion, :global-declarations, allows the user to make all symbol declarations and definitions
global to the assertion stack. In other words, when that option is enabled, declarations and
definitions become permanent, as opposed to being added to the assertion stack. Popping a

〈sort_dec〉 ::= ( 〈symbol〉 〈numeral〉 )
〈selector_dec〉 ::= ( 〈symbol〉 〈sort〉 )
〈constructor_dec〉 ::= ( 〈symbol〉 〈selector_dec〉∗ )
〈datatype_dec〉 ::= ( 〈constructor_dec〉+ ) | ( par ( 〈symbol〉+ ) ( 〈constructor_dec〉+ ) )

〈function_dec〉 ::= ( 〈symbol〉 ( 〈sorted_var〉∗ ) 〈sort〉 )
〈function_def 〉 ::= 〈symbol〉 ( 〈sorted_var〉∗ ) 〈sort〉 〈term〉
〈prop_literal〉 ::= 〈symbol〉 | ( not 〈symbol〉 )
〈command〉 ::= ( assert 〈term〉 )

| ( check-sat )
| ( check-sat-assuming ( 〈prop_literal〉∗ ) )
| ( declare-const 〈symbol〉 〈sort〉 )
| ( declare-datatype 〈symbol〉 〈datatype_dec〉)
| ( declare-datatypes ( 〈sort_dec〉n+1 ) ( 〈datatype_dec〉n+1 ) )
| ( declare-fun 〈symbol〉 ( 〈sort〉∗ ) 〈sort〉 )
| ( declare-sort 〈symbol〉 〈numeral〉 )
| ( define-fun 〈function_def 〉 )
| ( define-fun-rec 〈function_def 〉 )
| ( define-funs-rec ( 〈function_dec〉n+1 ) ( 〈term〉n+1 ) )
| ( define-sort 〈symbol〉 ( 〈symbol〉∗ ) 〈sort〉 )
| ( echo 〈string〉 )
| ( exit )
| ( get-assertions )
| ( get-assignment )
| ( get-info 〈info_flag〉 )
| ( get-model )
| ( get-option 〈keyword〉 )
| ( get-proof )
| ( get-unsat-assumptions )
| ( get-unsat-core )
| ( get-value ( 〈term〉+ ) )
| ( pop 〈numeral〉 )
| ( push 〈numeral〉 )
| ( reset )
| ( reset-assertions )
| ( set-info 〈attribute〉 )
| ( set-logic 〈symbol〉 )
| ( set-option 〈option〉 )

〈script〉 ::= 〈command〉∗

Figure 3.6: SMT-LIB Commands.

〈b_value〉 ::= true | false

〈option〉 ::= :diagnostic-output-channel 〈string〉
| :global-declarations 〈b_value〉
| :interactive-mode 〈b_value〉
| :print-success 〈b_value〉
| :produce-assertions 〈b_value〉
| :produce-assignments 〈b_value〉
| :produce-models 〈b_value〉
| :produce-proofs 〈b_value〉
| :produce-unsat-assumptions 〈b_value〉
| :produce-unsat-cores 〈b_value〉
| :random-seed 〈numeral〉
| :regular-output-channel 〈string〉
| :reproducible-resource-limit 〈numeral〉
| :verbosity 〈numeral〉
| 〈attribute〉

Figure 3.7: Command options.

〈info_flag〉 ::= :all-statistics | :assertion-stack-levels | :authors
| :error-behavior | :name | :reason-unknown
| :version | 〈keyword〉

Figure 3.8: Info flags.

stack level then has only the effect of removing asserted formulas (those in that level). Global
declarations and definitions can be removed only by a reset command.

Declared/defined symbols. Sort and function symbols introduced with a declaration or a
definition cannot have a name that begins with a dot (.), as such names are reserved for
solvers’ use, or with @, as such symbols are reserved for solver-defined abstract values.

3.9.1 Command responses

The possible responses that a solver can produce in response to commands are shown in
Figure 3.9. Every response must be an instance of 〈general_response〉 which specifies generic
response possibilities as well as command-specific responses for certain commands (specified
by 〈specific_success_response〉). In addition, with the :print-success option set to false, a
solver produces no response in cases when it would have otherwise returned success.

Regular output, including error messages, is printed on the regular output channel. Di-
agnostic output, including warnings and progress information, is printed on the diagnostic
output channel. These may be set using set-option and the corresponding attributes: re-
spectively, :regular-output-channel and :diagnostic-output-channel. The values of these
attributes should be (double-quote delimited) file names in the format specified by the POSIX
standard.12 The string literals "stdout" and "stderr" are reserved to refer specially to the

12This is the usual format adopted by all Unix-based operating systems, with / used as a separator for

〈error-behavior〉 ::= immediate-exit | continued-execution
〈reason-unknown〉 ::= memout | incomplete | 〈s_expr〉
〈model_response〉 ::= ( define-fun 〈function_def 〉 ) | ( define-fun-rec 〈function_def 〉 )

| ( define-funs-rec ( 〈function_dec〉n+1 ) ( 〈term〉n+1 ) )

〈info_response〉 ::= :assertion-stack-levels 〈numeral〉
| :authors 〈string〉
| :error-behavior 〈error-behavior〉
| :name 〈string〉
| :reason-unknown 〈reason-unknown〉
| :version 〈string〉
| 〈attribute〉

〈valuation_pair〉 ::= ( 〈term〉 〈term〉 )
〈t_valuation_pair〉 ::= ( 〈symbol〉 〈b_value〉 )

〈check_sat_response〉 ::= sat | unsat | unknown

〈echo_response〉 ::= 〈string〉
〈get_assertions_response〉 ::= ( 〈term〉∗ )
〈get_assignment_response〉 ::= ( 〈t_valuation_pair〉∗ )
〈get_info_response〉 ::= ( 〈info_response〉+ )

〈get_model_response〉 ::= ( 〈model_response〉∗ )
〈get_option_response〉 ::= 〈attribute_value〉
〈get_proof_response〉 ::= 〈s_expr〉
〈get_unsat_assump_response〉 ::= ( 〈symbol〉∗ )
〈get_unsat_core_response〉 ::= ( 〈symbol〉∗ )
〈get_value_response〉 ::= ( 〈valuation_pair〉+ )

〈specific_success_response〉 ::= 〈check_sat_response〉 | 〈echo_response〉
| 〈get_assertions_response〉 | 〈get_assignment_response〉
| 〈get_info_response〉 | 〈get_model_response〉
| 〈get_option_response〉 | 〈get_proof_response〉
| 〈get_unsat_assumptions_response〉
| 〈get_unsat_core_response〉 | 〈get_value_response〉

〈general_response〉 ::= success | 〈specific_success_response〉
| unsupported | ( error 〈string〉 )

Figure 3.9: Command responses.

(set-info :smt-lib-version 2.6)
; success

(set-logic QF_LIA)
; success

(declare-const w Int)
; success

(declare-const x Int)
; success

(declare-const y Int)
; success

(declare-const z Int)
; success

(assert (> x y))
; success

(assert (> y z))
; success

(set-option :print-success false)

(push 1)

(assert (> z x))

(check-sat)
; unsat

(get-info :all-statistics)
; (:time 0.01 :memory 0.2)

(pop 1)

(push 1)

(check-sat)
; sat

(exit)

Figure 3.10: Example script (over two columns), with expected solver responses in comments.

corresponding standard process channels (as opposed to disk files with that name).

Specific Responses. Specific responses are defined, in Figure 3.9, for the following com-
mands:

〈check_sat_response〉 for check-sat and check-sat-assuming
〈echo_response〉 for echo,
〈get_assertions_response〉 for get-assertions,
〈get_assignment_response〉 for get-assignment,
〈get_info_response〉 for get-info,
〈get_model_response〉 for get-model,
〈get_option_response〉 for get-option,
〈get_proof_response〉 for get-proof,
〈get_unsat_assump_response〉 for get-unsat-assumptions,
〈get_unsat_core_response〉 for get-unsat-core,
〈get_value_response〉 for get-value.

See Chapter 4 for more details.

3.9.2 Example scripts

We demonstrate some allowed behavior of a hypothetical solver in response to an example
script. Each command is followed by example legal output from the solver in a comment, if

(sub)directories, etc.

(set-info :smt-lib-version 2.6)
...
(set-option :print-success false)
(set-option :produce-models true)

(declare-const x Int)
(declare-const y Int)
(declare-fun f (Int) Int)

(assert (= (f x) (f y)))
(assert (not (= x y)))

(check-sat)
; sat

(get-value (x y))
; ((x 0)
; (y 1)
; )

(declare-const a (Array Int (List Int)))

...

(check-sat)
; sat

(get-value (a))
; ( (a (as @array1 (Array Int (List Int))))
; )

(get-value ((select @array1 2)))
; (((select (as @array1 (Array Int (List Int))) 2)
; (as @list0 (List Int))
; )
; )

(get-value ((first @list0) (rest @list0)))
; (((first (as @list0 (List Int))) 1)
; ((rest (as @list0 (List Int))) (as nil (List Int)))
; )

Figure 3.11: Another example script (excerpt), with expected solver responses in comments.

there is any. The script in Figure 3.10 makes two background assertions and then conducts
two independent queries. The get-info command requests information on the search using
the :all-statistics flag.13 The script in Figure 3.11 uses the get-value command to get
information about a particular model of the formula that the solver has reported satisfiable.

3.9.3 SMT-LIB Benchmarks

Starting with Version 2.0 of the SMT-LIB language, there is no explicit syntactic category of
benchmarks. Instead, meta-level information about a script used as a benchmark is included
in the script via the set-info command.

Benchmarks in the official SMT-LIB repository at www.smt-lib.org must satisfy additional
requirements on the meta-level information they contain and the order in which it appears.
Specifically, every benchmark must use set-info to set the attributes below as follows:

• :smt-lib-version, :source, :license, and :category must be set exactly once,

• :status must be set as many times as needed so that each occurrence of the command
check-sat-assuming or check-sat in the benchmark is preceded (not necessarily imme-
diately) by a corresponding :status info.14

Moreover, the set set-info call for attribute :smt-lib-version must be the very first com-
mand in the benchmark.

13Since the output of (get-info :all-statistics) is solver-specific, the response reported in the script is for
illustration purposes only.

14The same call to set-info can be used to provide the status for more than one call to check-sat-assuming or
check-sat, if that status is the same.

http://www.smt-lib.org

CHAPTER 4

Operational Semantics of SMT-LIB

This chapter specifies how a human user or a software client can interact with an SMT-LIB-
compliant solver. We do that by providing, as precisely as possible, an operational semantics
of SMT-LIB scripts, together with additional requirements on the input/output behavior of
the solver.

The expected interaction mode with a compliant solver is that of a read-eval-print loop:
the user or client application issues a command in the format of the Command Language
to the SMT solver via the solver’s standard textual input channel; the solver then responds
over two textual output channels, one for regular output and one for diagnostic output, and
waits for another command. A non-interactive mode is also allowed where the solver reads
commands from a script stored in a file. However, the solver’s output behavior should be
exactly the same as if the commands in the script had been sent to it one at a time.

Note that the primary goal of the SMT-LIB standard is, first and foremost, to support
convenient interaction with other programs, not human interaction. This has some influence
on the design of the command language.

There are other commands one might wish for an SMT solver to support beyond those
adopted here. In general, it is expected that time and more experience with the needs of
applications will drive the addition of further commands in later versions.

4.1 General Requirements

The command language contains commands for managing a stack of assertion levels and for
making queries about them. It includes commands to:

• declare and define new sort and function symbols (declare and define commands),

• add formulas to the current assertion level,

Start mode Assert mode

Sat mode

Unsat mode

e, gsio, r, ra

sl

ad, e, g,
gsio, p, ra

c

c
r

c, e, g, gamv, gsio
r

ad, p, ra

c

c, e, g, gsio, gpu

r

ad, p, ra

c

Command name abbreviations:

ad = assert, declare-*, define-*
c = check-sat*
e = echo
g = get-assertions

gamv = get-assignment, get-model, get-value
gsio = get-info, get-option, set-info, set-option

gpu = get-proof, get-unsat-*
p = pop, push
r = reset

ra = reset-assertions
sl = set-logic

Figure 4.1: Abstract view of transitions between solver execution modes. The symbol * here
stands for the matching wildcard.

• reset the assertion stack or the whole solver,

• push and pop assertion levels,

• check the joint satisfiability of all formulas in the assertion stack, possibly under addi-
tional assumptions (check commands),

• obtain further information following a check command (e.g., model information),

• set values for standard and solver-specific options,

• get standard and solver-specific information from the solver.

This section provides some background and general requirements on how these function-
alities are to be supported. The next section, Section 4.2 provides more details on how each
compliant solver is to execute each command.

4.1.1 Execution Modes

At a high-level, a compliant solver can be understood as being at all times in one of four
execution modes: a start mode, an assert mode and two query modes, sat and unsat. The solver

starts in start mode, moves to assert mode once a logic is set, and then moves to one of the
two query modes after executing a check command. Any command other than reset that
modifies the assertion stack brings the solver back from a query mode to the assert mode. The
reset command takes the solver back to start mode.

The transition system in Figure 4.1 illustrates in some detail which commands can trigger
which mode transitions. The set of labels for each transition describes the commands that
may cause it. With the exception of exit, if a command does not appear on any transitions
originating from a mode, it is not permitted in that mode. The exit command, which causes
the solver to quit, can be issued in any mode.

The solver must respond with an error when given a command not permitted in the
current mode. Because of its level of abstraction, the transition system diagram in Figure 4.1
does not specify the conditions under which a specific command causes a transition to one
mode as opposed to another; see Section 4.2 for details on that. Similarly, the diagram does
not account for the fact that some specific options can be set only in certain modes. Such
restrictions are described in Section 4.1.7.

4.1.2 Solver responses

Regular output, including responses and errors, produced by compliant solvers should be
written to the regular output channel. Diagnostic output, including warnings, debugging,
tracing, or progress information, should be written to the diagnostic output channel. These
channels may be set with the 〈set-option〉 command (see Section 4.1.7 below). By default they
are the standard output and standard error channels, respectively.

Generally, once a solver completes its processing in response to a command, it should
print to its regular output channel a 〈general_response〉:

〈general_response〉 ::= success | 〈specific_success_response〉
| unsupported | ( error 〈string〉 )

The value success is the default response for a successful execution of a supported command.
A number of commands have a more detailed response in place of success, discussed in Sec-
tion 4.2 for each of them. The value unsupported should be returned if the command or some
specific input to it is not supported by the solver. An expression of the form (error e) should
be returned for any kind of error situation (wrong command syntax, incorrect parameters, er-
roneous execution, and so on). The value of e is a solver-specific string containing a message
that describes the problem encountered.1

Any response which is not double-quoted and not parenthesized must be followed by at
least one whitespace character (for example, a new line character).(25)

Several options described in Section 4.1.7 below affect the printing of responses, in partic-
ular by suppressing the printing of success, or by redirecting the regular or diagnostic output
channels.

1Returning the empty string is allowed but discouraged because of its uninformative content.

Errors and solver state. Solvers have two options when encountering errors. For both op-
tions, they first print an error message in the 〈general_response〉 format. Then, they may
either immediately exit with a non-zero exit status, or continue accepting commands. In the
second case, the solver’s state remains unmodified by the error-generating command, except
possibly for timing and diagnostic information. In particular, the assertion stack, discussed in
Section 4.1.4, is unchanged.(26)

The predefined :error-behavior attribute can be used with the get-info command to
check which error behavior the tool supports (see Section 4.1.8 below).

4.1.3 Printing of terms and defined symbols

Several commands request the solver to print sets of terms. While some commands, naturally,
place additional semantic requirements on these sets, the general syntactic requirement is that
output terms must be well-sorted with respect to the current signature (as defined below in
Section 4.1.4).

All output from a compliant solver should print any symbols defined with define-sort
and define-fun just as they are, without replacing them by the expression they are defined to
be equal to. This approach generally keeps the output from solvers much more compact than
it would be if definitions were expanded.

4.1.4 The assertion stack

A compliant solver maintains a stack of sets, each of which consists of assertions. Assertions are
formulas, declarations, and definitions. We will use the following terminology with regards
to this data structure:

• assertion stack: the single stack of sets of assertions;

• assertion level: an element of the assertion stack (i.e., a set of assertions);

• context: the union of all the assertion levels on the assertion stack together with any
global declarations (see Section 4.1.5);

• current assertion level: the assertion level at the top of the stack (i.e., the most recent);

• first assertion level: the first assertion level in the stack (i.e., the least recent);

• current signature: the signature determined by the logic specified with the most recent
set-logic command and by the set of sort symbols and rank associations (for function
symbols) in the current context.

Initially, when the solver starts, the assertion stack consists of a single element, the first
assertion level, which is empty. While new assertions can be added to this set, the set itself
cannot be removed from the stack with a pop operation. The following commands modify
the current context:

assert, declare-sort, declare-fun, declare-const, define-sort, define-fun,
define-fun-rec, define-funs-rec, pop, push, reset, and reset-assertions.

4.1.5 Symbol declarations and definitions

A number of commands allow the declaration or definition of a function or sort symbol. By
default, these declarations and definitions are added to the current assertion level when the
corresponding command is executed. Popping that assertion level removes them.(27) As an
alternative, declarations and definitions can all be made global by running the solver with the
option :global-declarations set to true. When running with this option set, all declarations
and definitions become permanent. That is, they survive any pop operations on the assertion
stack as well as invocations of reset-assertions and can only be removed by a global reset,
achieved with the reset command.(28)

Well-sortedness checks, required for commands that use sorts or terms, are always done
with respect to the current signature. It is an error to declare or define a symbol that is already
in the current signature. This implies in particular that, contrary to theory function symbols,
user-defined function symbols cannot be overloaded.(29)

4.1.6 In-line definitions

Any closed subterm t occurring in the argument(s) of a command c can be optionally an-
notated with a :named attribute; that is, it can appear as (! t :named f) where f is a fresh
function symbol from 〈symbol〉. For such a command c, let

(! t1 :named f1), . . . , (! tn :named fn)

be the in-order enumeration of all the named subterms of c. The semantics of the command c
is the same as the sequence of commands

(define-fun f1 () σ1 t′1)
...
(define-fun fn () σn t′n)
c′

where, for each i = 1, . . . , n, (i) σi is the sort of ti with respect to the current signature up to
the declaration of fi, (ii) t′i is the term obtained from ti by removing all its :named annotations,
and (iii) c′ is similarly obtained from c by removing all its :named annotations.

By these semantics, each label fi can occur, as a constant symbol, in any subexpression of c
that comes after (! ti :named fi) in the in-order traversal of c, as well as after the command
c itself. The labels f1, . . . , fn can be used like any other user-defined nullary function symbols,
with the same visibility and scoping restrictions they would have if they had been defined
with the sequence of commands above. However, contrary to function symbols introduced
by :define-fun, labels have an additional, dedicated use in the commands :get-assignment
and :get-unsat-core (see Section 4.2).

4.1.7 Solver options

Solver options may be set using the set-option command, and their current values can be ob-
tained using the get-option command. If a solver does not support the setting of a particular
option, for either command it should output unsupported.

Solver-specific option names are allowed and indeed expected. A set of standard options
is presented in this subsection; refer to Figure 3.7 for their format. We discuss each option
below, specifying also their default values and whether or not compliant solvers are required
to support them. It is understood that if a solver does not support one of the optional standard
options below, it behaves as if that option was permanently set to its default value.

Some options can be set only when the solver is in start mode. We list the mode when
that is the case. Attempting to set an option when the solver is not in a permitted mode
should trigger an error response. Each option starting with the produce- prefix is a Boolean
option that enables a specific command. When such an option is set to false, calling the
corresponding command should trigger an error response.

The set of standard options is likely to be expanded or otherwise revised as further desir-
able common options and kinds of information across tools are identified.

:diagnostic-output-channel default: "stderr" support: required
The argument is a string consisting of the name of a file to be used subsequently as the
diagnostic output channel. The input value "stderr" is interpreted specially to mean
the solver’s standard error channel. With other filenames, subsequent solver output is to
be appended to the named file (and the file should be first created if it does not already
exist).

:global-declarations default: false support: optional mode: start
If the solver supports this option, setting it to true causes all declarations and definitions
to be global (permanent) as opposed to being added to the current assertion level.

:interactive-mode default: false support: optional mode: start
The old name for produce-assertions. Deprecated.

:print-success default: true support: required
Setting this option to false causes the solver to suppress the printing of success in all
responses to commands. Other output remains unchanged.

:produce-assertions default: false support: optional mode: start
If the solver supports this option, setting it to true enables the get-assertions com-
mand. This option was called interactive-mode in previous versions.

:produce-assignments default: false support: optional mode: start
If supported, this enables the command get-assignment.

:produce-models default: false support: optional mode: start
If supported, this enables the commands get-value and get-model.

:produce-proofs default: false support: optional mode: start
If supported, this enables the command get-proof.

:produce-unsat-assumptions default: false support: optional mode: start
If supported, this enables the command get-unsat-assumptions.

:produce-unsat-cores default: false support: optional mode: start
If supported, this enables the command get-unsat-core.

:random-seed default: 0 support: optional mode: start
The argument is a numeral for the solver to use as a random seed, in case the solver
uses (pseudo-)randomization. The default value of 0 means that the solver can use
any random seed—possibly even a different one for each run of the same script. The
intended use of the option is to force the solver to produce identical results whenever
given identical input (including identical non-zero seeds) on repeated runs of the solver.

:regular-output-channel default: "stdout" support: required
The argument should be a filename to use subsequently for the regular output channel.
The input value "stdout" is interpreted specially to mean the solver’s standard output
channel. With other filenames, subsequent solver output is to be appended to the named
file (and the file should be first created if it does not already exist).

:reproducible-resource-limit default: 0 support: optional
If the solver supports this option, setting it to 0 disables it. Setting it a non-zero numeral
n will cause each subsequent check command to terminate within a bounded amount
of time dependent on n. The internal implementation of this option and its relation
to run time or other concrete resources can be solver-specific. However, it is required
that the invocation of a check command return unknown whenever the solver is unable
to determine the satisfiability of the formulas in the current context within the current
resource limit. Setting a higher value of n should allow more resources to be used,
which may cause the command to return sat or unsat instead of unknown. Furthermore,
the returned result should depend deterministically on n; specifically, it should be the
same every time the solver is run with the same sequence of previous commands on the
same machine (and with an arbitrarily long external time out). If the solver makes use
of randomization, it may require the :random-seed option to be set to a value other than
0 before :reproducible-resource-limit can be set to a positive value.(30)

:verbosity default: 0 support: optional
The argument is a numeral controlling the level of diagnostic output produced by the
solver. All such output should be written to the diagnostic output channel(31) which can
be set and later changed via the diagnostic-output-channel option. An argument of 0
requests that no such output be produced. Higher values correspond to more verbose
output.

4.1.8 Solver information

The format for responses to get-info commands, both for standard and solver-specific in-
formation flags, is defined by the 〈get_info_response〉 category in Figure 3.9. The standard
get-info flags and specific formats for their corresponding responses are given next.

:all-statistics support: optional mode: sat, unsat
Solvers reply with a parenthesis-delimited sequence of 〈info_response〉 values (see Fig-
ure 3.9) providing various statistics on the execution of the most recent check command.

No standard statistics are defined for the time being,(32) so they are all solver-specific.
Executions of get-info with :all-statistics are allowed only when the solver is in
sat or unsat mode.

:assertion-stack-levels support: optional
The response is a pair of the form (:assertion-stack-levels n) where n is a numeral
indicating the current number of levels in the assertion stack besides the first assertion
level.(33)

:authors support: required
The response is a pair of the form (:authors s) where s is a string literal listing the
names of the solver’s authors.

:error-behavior support: required
The response is a pair of the form (:error-behavior r) where r is either immediate-exit
or continued-execution. A response of immediate-exit indicates that the solver will
exit immediately when an error is encountered. A response of continued-execution
indicates that when an error is encountered, the solver will return to the state it was
in immediately before the command triggering the error, and continue accepting and
executing new commands. See Section 4.1.2 for more information.

:name support: required
The response is a pair of the form (:name s) where s is a string literal with the name of
the solver.

:reason-unknown support: optional mode: sat
Executions of get-info with :reason-unknown are allowed only when the solver is in sat
mode following a check command whose response was unknown. The response is a pair
of the form (:reason-unknown r) where r is an element of 〈reason_unknown〉 giving a
short reason why the solver could not successfully check satisfiability. In general, this
reason can be provided by a solver-defined s-expression. Two predefined s-expressions
are memout, for out of memory, and incomplete, which indicates that the solver knows
it is incomplete for the class of formulas containing the most recent check query.

:version support: required
The response is a string literal with the version number of the solver (e.g., "1.2").

4.2 Commands

The full set of commands and their expected behavior are described in this section. Com-
mands may impose restrictions on their arguments as well as restrictions on when they can
be issued. Unless otherwise specified, the solver is required to produce an error when any of
these restrictions is violated. Figure 4.1 describes the commands permitted in each execution
mode and the mode transitions each command may trigger. We specify below the conditions
under which a command triggers one mode transition versus another only for commands that
may trigger more than one transition.

4.2.1 (Re)starting and terminating

(reset) resets the solver completely to the state it had after it was started and before it started
reading commands.(34)

(set-logic l) tells the solver what logic, in the sense of Section 5.5, is being used. The argu-
ment l can be the name of a logic in the SMT-LIB catalog or of some other solver-specific
logic. The effect of the command is to add globally (and permanently) a declaration of
each sort and function symbol in the logic.

The argument l can also be the predefined symbol ALL. With this argument, the solver
sets the logic to the most general logic it supports.(35) Note that while the reaction
to (set-logic ALL) is the same for every compliant solver, the chosen logic is solver-
specific.

We refer to the logic set by the most recent set-logic command as the current logic.

(set-option o v) sets a solver’s option o to a specific value v. More details on predefined
options and required behavior are provided in Section 4.1.7. In general, if a solver does
not support the setting of a particular option, its response to this command should be
unsupported. If the option is one of the predefined ones it should also leave it unchanged
from its default value. The effect of setting a supported option is immediate. In partic-
ular, for options that affect the solver’s output, such as :diagnostic-output-channel,
:regular-output-channel and :print-success, the effect applies already to the output
of the very command that is setting the option.

Note that some of the options defined in Section 4.1.7 may only be set in start mode.(36)

(exit) instructs the solver to exit.

4.2.2 Modifying the assertion stack

(push n) pushes n empty assertion levels onto the assertion stack.2 If n is 0, no assertion
levels are pushed.

(pop n) where n is smaller than the number of assertion levels in the stack, pops the n most-
recent assertion levels from the stack.3 Note that the first assertion level, which is not
created by a push command, cannot be popped.

(reset-assertions) removes from the assertion stack all assertion levels beyond the first
one. In addition, it removes all assertions from the first assertion level. Declarations
and definitions resulting from the set-logic command are unaffected (because they are
global). Similarly, if the option :global-declarations has value true at the time the
command is executed, then all declarations and definitions remain unaffected. Note that
any information set with set-option commands is preserved in any case.

2Typically, n = 1.
3When n is 0, no assertion levels are popped.

4.2.3 Introducing new symbols

The next seven commands allow one to introduce new sort or function symbols by providing
them with a rank declaration (declare-sort, declare-fun and declare-const) or also with
a definition (define-sort, define-fun, define-fun-rec and define-funs-rec). We refer to
the former as user-declared symbols and the latter as user-defined symbols. Declarations and
definitions are made global (permanent) or are added to the current assertion level depending
on whether the :global-declarations option is set to true or not.

(declare-sort s n) adds sort symbol s with associated arity n. It is an error if s is a sort
symbol present in the current signature.

(define-sort s (u1 · · · un) τ) with n ≥ 0 adds sort symbol s with associated arity n. Sub-
sequent well-sortedness checks must treat a sort term like (s σ1 · · · σn) as an abbrevi-
ation for the term obtained by simultaneously substituting σi for ui, for i ∈ {1, . . . , n}, in
τ.(37)

The command reports an error if s is a sort symbol already present in the current sig-
nature or if τ is not a well-defined parametric sort with respect to the current signature.
This restriction prohibits (meaningless) circular definitions where τ contains s.

(declare-fun f (σ1 · · · σn) σ) with n ≥ 0 adds a new symbol f with associated rank
σ1 · · · σnσ. The command reports an error if a function symbol with name f is already
present in the current signature.

(declare-const f σ) abbreviates the command (declare-fun f () σ).

(declare-datatypes ((δ1 k1) · · · (δn kn)) (d1 · · · dn)) with n > 0 introduces n alge-
braic datatypes δ1, . . . , δn with respective arities k1, . . . , kn and declarations d1, . . . , dn.
Let δ = δi, k = ki and d = di for i ∈ {1, . . . , n}. If k > 0 then d is an expression of the
form (par (u1 · · · uk) l) where u1, . . . , uk are sort parameters; otherwise, it is just l. In
either case, l is a (parenthesis-delimited) list of one or more expressions of the form

(c (s1 τ1) · · · (sm τm))

where (i) m ≥ 0, (ii) c is a symbol, a constructor for δ, and (iii) for each j = 1, . . . , m,
sj is a symbol, a selector for c, and (iv) τj is a sort term that contains no occurrences of
δ1, . . . , δn below its top symbol. In the parametric case, the terms τi can contain the sort
parameters u1, . . . , uk. The datatype δ must be well founded in the following inductive
sense: it must have a constructor of rank τ1 · · · τmδ such that τ1 · · · τm does not contain
any of the datatypes from {δ1, . . . , δn} or, if it does contain some, they are well founded.

A compliant solver must return an error in response to invocations of this command
that do not satisfy all of the restrictions above.

In the parametric case, the command has the effect of declaring each δ as a sort symbol
of arity k; each constructor c as a function symbol of parametric rank τ1 · · · τmδ; and each
si as a function symbol of parametric rank δτi. The non-parametric case is analogous.

Note that the sort terms τ1, . . . , τm can contain any previously defined sort symbol as
well as any of the datatypes δ1, . . . , δn, as long as those datatypes are well founded.(38)

This allows the declaration of recursive and mutually recursive datatypes.(39)

On successfully executing this command, for each constructor c in a declared datatype
δ, the solver will also automatically declare a tester with rank δ Bool. The tester’s name
is an indexed identifier (see Section 3.3) of the form (_ is c).

; an enumeration datatype
(declare -datatypes ( (Color 0) ) (

( (red) (green) (blue) ))
)
; testers: (_ is red), (_ is green)

; integer lists with "empty" and "insert" constructors
(declare -datatypes ( (IntList 0) ) (

( (empty) (insert (head Int) (tail IntList) )))
)
; testers: (_ is empty), (_ is insert)

; parametric lists with "nil" and "cons" constructors
(declare -datatypes ( (List 1) ) (

(par (T) ( (nil) (cons (car T) (cdr (List T)) )))))

; option datatype
(declare -datatypes ( (Option 1) ) (

(par (X) ( (none) (some (val X)) ))))

; parametric pairs
(declare -datatypes ( (Pair 2) ) (

(par (X Y) ( (pair (first X) (second Y)) ))))

; two mutually recursive datatypes
(declare -datatypes ( (Tree 1) (TreeList 1) ) (

; Tree
(par (X) ( (node (value X) (children (TreeList X)) )))
; TreeList
(par (Y) ( (empty)

(insert (head (Tree Y)) (tail (TreeList Y))) ))))

Since δ1, . . . , δn are sort symbols, none of them can be a previously declared sort symbol.
Similarly, constructors and selectors are function symbols, so none of them can be a
previous declared/defined function symbol. This has the effect of also prohibiting, for
instance, the use of the same constructor in different datatypes or the use of repeated
instances of the same selector in the same datatype.(40)

(declare-datatype δ d) is an abbreviation of

(declare-datatypes ((δ 0)) (d))

if δ is not parametric, and an abbreviation of

(declare-datatypes ((δ k)) (d))

if d has the form (par (u1 · · · uk) l). This command provides a simpler syntax for
defining a single datatype.

; an enumeration datatype
(declare -datatype Color ( (red) (green) (blue) ))

(declare -datatype IntList
( (empty)

(insert (head Int) (tail IntList) )))

(declare -datatype List (par (E)
( (nil)

(cons (car E) (cdr (List E)) ))))

(declare -datatype Option (par (X)
( (none)

(some (val X) ))))

(declare -datatype Pair (par (X Y)
( (pair (first X) (second Y)) ))))

(define-fun f ((x1 σ1) · · · (xn σn)) σ t) with n ≥ 0 and t not containing f is semanti-
cally equivalent to the command sequence

(declare-fun f (σ1 · · · σn) σ)
(assert (forall ((x1 σ1) · · · (xn σn)) (= ( f x1 · · · xn) t)).

Note that the restriction on t prohibits recursive or mutually recursive definitions, which
are instead provided by define-fun-rec and define-funs-rec. The command reports
an error if a function symbol with name f is already present in the current signature or
if the argument t is not a well-sorted term of sort σ with respect to the current signature
extended with the sort associations (x1 : σ1), . . . , (xn : σn).

(define-funs-rec (d1 · · · dm) (t1 · · · tm)), where m > 0 and for i = 1, . . . , m, di has the
form

( fi ((xi,1 σi,1) · · · (xi,ni σi,ni)) σi)

with ni ≥ 0 and f1, . . . , fm pairwise distinct, is semantically equivalent to the command
sequence

(declare-fun f1 (σ1,1 · · · σ1,n1) σ1)
...

(declare-fun fm (σm,1 · · · σm,nm) σm)
(assert (forall ((x1,1 σ1,1) · · · (x1,n1 σ1,n1)) (= ( f1 x1,1 · · · x1,n1) t1))

...
(assert (forall ((xm,1 σm,1) · · · (xm,nm σm,nm)) (= ( fm xm,1 · · · xm,nm) tm)) .

This command can be used to define multiple functions recursively, in particular, mu-
tually recursively.(41) Mutual recursion is possible since each term ti can contain any
applications of f1, . . . , fm.

Note that, according to the semantics above, define-funs-rec imposes no requirements
that each fi be terminating (a meaningless notion in our context) or even well-defined.4

The only requirement is on the well-sortedness of the definitions.

The command reports an error if for any i ∈ {1, . . . , m} a function symbol with name fi
is already present in the current signature or if ti is not a well-sorted term of sort σi with
respect to the current signature extended with the sort associations ( f1 : σ1,1 · · · σ1,n1 σ1),
. . . , ( fm : σm,1 · · · σm,nm σm) and (xi,1 : σi,1), . . . , (xi,ni : σi,ni).

(define-fun-rec f ((x1 σ1) · · · (xn σn)) σ t) is an abbreviation of

(define-funs-rec (( f ((x1 σ1) · · · (xn σn)) σ)) (t))

It provides a simpler syntax to define individual recursive functions.

4.2.4 Asserting and inspecting formulas

(assert t) where t is a well-sorted formula (i.e., a well-sorted term of sort Bool), adds t to the
current assertion level. The well-sortedness requirement is with respect to the current
signature.

Instances of this command of the form (assert (! t :named f)), where the asserted
formula t is given a label f , have the additional effect of adding t to the formulas tracked
by the commands get-assignment and get-unsat-core, as explained later.

(get-assertions) causes the solver to print the current set of all asserted formulas as a
sequence of the form ( f1 · · · fn). Each fi is a formula syntactically identical, modulo
whitespace, to one of the formulas entered with an assert command and currently in
the context. Solvers are not allowed to print formulas equivalent to or derived from the
asserted formulas.(42)

The command can be issued only if the :produce-assertions option, which is set to
false by default, is set to true (see Section 4.1.7).

4In fact, it is even possible, although certainly not desirable, to have a definition like (define-funs-rec ((f
(x Bool) Bool)) ((not (f x))) ), which makes the set of formulas in the context unsatisfiable.

4.2.5 Checking for satisfiability

(check-sat) is an abbreviation of (check-sat-assuming ()).

(check-sat-assuming (a1 · · · an)) where n ≥ 0 and a1, · · · , an are terms of sort Bool, in-
structs the solver to check whether the conjunction of all the formulas in the current
context and the assumptions a1, · · · , an is satisfiable in the extension of the current logic
with all the current user-declared and user-defined symbols. The assumptions a1, . . . , an

can only have the form p or (not p) where p is a user-declared/defined Boolean con-
stant, i.e., a nullary function symbol of sort Bool.(43)

Conceptually, this command asks the solver to search for a model of the logic that satis-
fies all the currently asserted formulas as well as the current assumptions. When it has
finished attempting to do this, the solver should reply on its regular output channel (see
Section 4.1.2) using the response format defined by 〈check_sat_response〉 in Figure 3.9.
A sat response indicates that the solver has found a model, an unsat response that
the solver has established there is no model, and an unknown response that the search
was inconclusive—because of resource limits, solver incompleteness, or other reasons.
On reporting sat or unknown the solver should move to sat mode—and then respond to
get-assignment, get-model, and get-value commands provided that the corresponding
enabling option is set to true. On reporting unsat, it should move to unsat mode—and
then respond to get-proof, get-unsat-assumptions, and get-unsat-core commands
provided that the corresponding enabling option is set to true.

Regardless of how it is implemented internally, a check-sat-assuming command should
preserve the current context in the sense that at the end of the command’s execution the
context should be the same as it was right before the execution.

Note that a check-sat-assuming command can be issued also when the solver is already
in sat or unsat mode (in this case, the context is necessarily the same as for the previous
check command). However, it is possible for the solver to switch from sat to unsat mode
or vice versa if the latest command has a different set of assumptions from the previous
one.

4.2.6 Inspecting models

The next three commands can be issued only when the solver is in sat mode, and provide
information related to the most recent check command. In that case, the solver will have
identified a model A (as defined in Section 5.3) of the current logic, and produces responses
with respect to that same model until it receives the next check command or it exits the
sat mode, whichever comes first. The model A is required to satisfy all currently asserted
formulas and current assumptions only if the most recent check command reported sat.(44)

The internal representation of the model A is not exposed by the solver. Similarly to
an abstract data type, the model can be inspected only through the three commands below.
As a consequence, it can even be partial internally and extended as needed in response to
successive invocations of some of these commands.5

5In that case, of course, the solver has to be sure that its partial model can be indeed extended as needed.

(get-value (t1 · · · tn)) where n > 0 and each ti is a well-sorted closed quantifier-free term,
returns for each ti a value term vi

6 that is equivalent to ti in the current model A (see
above). Specifically, vi has the same sort as ti, and ti is interpreted the same way as vi
in A. The values are returned as a sequence of pairs of the form ((t1 v1) · · · (tn vn)).
The terms v1, . . . , vn are allowed to contain symbols not in the current signature only if
they are abstract values, i.e., constant symbols starting with the special character @.(45)

Since these are solver-defined, their sort is not known to the user. Therefore, additionally,
each occurrence of an abstract value a of sort σ in v1, . . . , vn has to be contained in a term
of the form (as a σ) which makes the sort explicit.

Note that the returned abstract values are used only to express information about the
current model A. They cannot be used in later assert commands since they are neither
theory symbols nor user-defined ones. However, they can be used in later get-value
commands on the same model.

There is no requirement that different permutations of the same set of get-value calls
produce the same value for the input terms. The only requirement is that syntactically
different values of the same sort returned by the solver have different meaning in the
model.7

The command can be issued only if the :produce-models option, which is set to false
by default, is set to true (see Section 4.1.7).

(get-assignment) can be seen as a light-weight and restricted version of get-value that asks
for a truth assignment for a selected set of previously entered formulas.(46)

The command returns a sequence of the form (( f1 b1) · · · ( fn bn)) with n ≥ 0. A pair
( fi bi) is in the returned sequence if and only if fi is the label of a (sub)term of the form
(! ti :named fi) in the context, with ti a closed term of sort Bool, and bi is the value
(true or false) that ti has in the current model A.

The command can be issued only if the :produce-assignments option, which is set to
false by default, is set to true (see Section 4.1.7).

(get-model) returns a list (d1 · · · dk) of definitions specifying all and only the current user-
declared function symbols {g1, . . . , gm} in the current model A. The interpretation of
each symbol is provided in exactly one of the definition d1,. . . dk. The define commands
d1,. . . dk have one of the following forms:(47)

• (define-fun f ((x1 σ1) · · · (xn σn)) σ t)
where n ≥ 0, f has rank σ1 · · · σnσ, t is a term not containing f , and the formula

(forall ((x1 σ1) · · · (xn σn)) (= ( f x1 · · · xn) t))

is well-sorted and satisfied by A. The term t is expected, although not required, to
be a value when f is a constant (i.e., when n = 0).

6Recall that value terms are particular ground terms defined in a logic for each sort (see Subsection 5.5.1).
7So, for instance, in a logic of rational numbers and values of the form (/ m n) and (/ (- m) n) with m, n

numerals, the solver cannot use both the terms (/ 1 3) and (/ 2 6) as output values for get-value.

• (define-funs-rec (( f1 ((x1,1 σ1,1) · · · (x1,n1 σ1,n1)) σ1) · · ·
( fp ((xp,1 σp,1) · · · (xp,np σp,np)) σp)) (t1 · · · tp))

where ni > 0 for i = 1, . . . , p, and the formula

(and (forall ((x1,1 σ1,1) · · · (x1,n σ1,n1)) (= ( f1 x1,1 · · · x1,n1) t1))
...

(forall ((xp,1 σp,1) · · · (xp,np σp,np)) (= ( fp xp,1 · · · xp,np) tp)))

is well-sorted and satisfied by A.

• (define-fun-rec f ((x1 σ1) · · · (xn σn)) σ t) where n > 0 and the formula

(forall ((x1 σ1) · · · (xn σn)) (= ( f x1 · · · xn) t))

is well-sorted and satisfied by A.

Similarly to the response of get-value, the terms t, t1, . . . , tm above are allowed to contain
symbols not in the current signature only if they are abstract values. Morevever, each
occurrence of an abstract value a of sort σ in t, t1, . . . , tm has to be contained in a term of
the form (as a σ).

Later versions of the standard may impose stronger requirements on the returned defi-
nitions. For now there is only an expectation that, when possible, the solver will provide
definitions that have a unique interpretation over the current signature.(48)

The command can be issued only if the :produce-models option, which is set to false
by default, is set to true (see Section 4.1.7).

4.2.7 Inspecting proofs

The next three commands can be issued only when the solver is in unsat mode, and provide
information related to the most recent check command (which produced an unsat response).

(get-unsat-assumptions) returns a subset a1, . . . , an of the assumptions in the most recent
check-sat-assuming command. These assumptions are such that issuing the command
(check-sat-assuming (a1 · · · an)) instead would have still produced an unsat re-
sponse. The returned sequence is not required to be minimal.(49)

The command can be issued only if the :produce-unsat-assumptions option, which is
set to false by default, is set to true (see Section 4.1.7).

(get-proof) asks the solver for a proof of unsatisfiability for the set of all formulas in the
current context. The command can be issued only if the most recent check command
had an empty set of assumptions. The solver responds by printing a refutation proof
on its regular output channel. The format of the proof is solver-specific.(50) The only
requirement is that, like all responses, it be a member of 〈s_expr〉.
The command can be issued only if the :produce-proofs option, which is set to false
by default, is set to true (see Section 4.1.7).

(get-unsat-core) asks the solver to identify an unsatisfiable core, a subset of all the formulas
in the current context that is unsatisfiable by itself. The command can be issued only
if the most recent check command had an empty set of assumptions. The solver selects
from the unsatisfiable core only those formulas that have been asserted with a command
of the form (assert (! t :named f)), and returns a sequence ( f1 · · · fn) of those
labels. Unlabeled formulas in the unsatisfiable core are simply not reported.(51)

The semantics of this command’s output is that the reported assertions together with all
the unlabeled ones in the set of all assertions are jointly unsatisfiable. In practice then,
not labeling assertions is useful for unsat core detection purposes only when the user is
sure that the set of all unlabeled assertions is satisfiable.

The command can be issued only if the :produce-unsat-cores option, which is set to
false by default, is set to true (see Section 4.1.7).

4.2.8 Inspecting settings

(get-info f) where f is an element of 〈info_flag〉 outputs solver information as specified
in Section 4.1.8. If a solver does not support a (standard or non-standard) flag f , it just
outputs unsupported.

(get-option o) outputs the current value of a solver’s option o as an element of 〈attribute_value〉.
The form of that value depends on the specific option. More details on standard options
and required behavior are provided in Section 4.1.7. If a solver does not support the
setting of a standard option o, the command outputs the option’s default value. For an
unsupported non-predefined option the command outputs unsupported.

4.2.9 Script information

(echo s) where s is a string literal, simply prints back s as is—including the surrounding
double-quotes.(52)

(set-info a) where a is an element of 〈attribute〉 has no effect on the assertion stack. Its
only purpose is to allow the insertion of structured meta information in a script.(53)

Typically then, a solver will just parse the command and do nothing with it, except for
printing a response (success or an error, for instance, if the argument is not an element
of 〈attribute〉).
There is only a small number of predefined set-info attributes, which are described
below together with their possible values. These attributes are used in particular in the
official SMT-LIB benchmarks at www.smt-lib.org.

:smt-lib-version possible values: a decimal.
The value of this attribute is the version of SMT-LIB used by the benchmark (e.g.,
2.6). For benchmarks in the official repository a call to set-info with this attribute
can occur only as the first command of a script.

http://www.smt-lib.org/benchmarks.shtml

:source possible values: a string or a quoted symbol.
The value of this attribute is a textual description of the benchmark’s source, con-
taining, as appropriate, such information as the name of person(s) who generated
the benchmark; the generation date; the tool that generated it; the intended ap-
plication; the solvers that were initially used or targeted to check the benchmarks;
references to related publications; any other information the benchmark author
deems useful.

:category possible values: "crafted", "random", and "industrial".
The value "crafted" indicates that the benchmark was hand-crafted while "random"
indicates that it was generated by a random process; "industrial" is reserved for
everything else.(54)

:license possible values: a string.
This is a description of the license under which the benchmark is distributed. It can
be the actual text of the license, or the URL of a web site containing the description.

:status possible values: sat, unsat, and unknown.
Each occurrence of the command (set-info :status sat) (respectively, (set-info
:status unsat)) indicates that the next check command in the script is expected
to return sat (respectively, unsat). More precisely, the expected value of a check
command in a script is the one indicated by the most recent command of the form
(set-info :status v) in the script. The value unknown is used when the expected
value is not known.(55)

Remark 7 (set-info and get-info are unrelated). Contrary to what their names might sug-
gest, set-info and get-info are not related. The first command is used to store information
about a script, the second to obtain solver-specific information.(56)



CHAPTER 5

Logical Semantics of SMT-LIB Formulas

The underlying logic of the SMT-LIB language is a variant of many-sorted first-order logic
(FOL) with equality [Man93, Gal86, End01], although it incorporates some syntactic features of
higher-order logics: in particular, the identification of formulas with terms of a distinguished
Boolean sort, and the use of sort symbols of arity greater than 0.

These features make for a more flexible and syntactically more uniform logical language.
However, while not exactly syntactic sugar, they do not change the essence of SMT-LIB logic
with respect to traditional many-sorted FOL. Quantifiers are still first-order, the sort structure
is flat (no subsorts), the logic’s type system has no function (arrow) types, no type quanti-
fiers, no dependent types, no provisions for parametric or subsort polymorphism. The only
polymorphism is of the ad-hoc variety (a function symbol can be given more than one rank),
although there is a syntactical mechanism for approximating parametric polymorphism. As a
consequence, all the classical meta-theoretic results from many-sorted FOL apply to SMT-LIB
logic when considered in its full generality, that is, with no restrictions on the possible mod-
els other than those imposed by the Core background theory introduced in Subection 3.7.1.
Those results still hold with recursively axiomatizable background theories, i.e., theories defined
as the set of all models of a recursive set of closed formulas (or axioms). As pointed out in
Section 3.7.1 the ability to use non-recursively axiomatizable theories as background theories
actually gives SMT-LIB logic the expressive power of higher-order logics. However, a formal
treatment of this aspect is beyond the scope of this document.

To define SMT-LIB logic and its semantics it is convenient to work with a more abstract
syntax than the concrete S-expression-based syntax of the SMT-LIB language. The formal
semantics of concrete SMT-LIB expressions is then given by means of a translation into this
abstract syntax. A formal definition of this translation might be provided in later releases of
this document. Until then, we will appeal to the reader’s intuition and on the fact that the
translation is defined as one would expect.

(Sorts) σ ::= s σ∗

(Parametric Sorts) τ ::= u | s τ∗

Figure 5.1: Abstract syntax for sort terms

The translation also maps concrete predefined symbols and keywords to their abstract
counterpart. To facilitate reading, usually the abstract version of a predefined concrete symbol
is denoted by the symbol’s name in Roman bold font (e.g., Bool for Bool). The same is done
for keywords (e.g., definition for :definition).

To define our target abstract syntax we start by fixing the following pairwise disjoint sets
of (abstract) symbols and values:

• an infinite set S of sort symbols s containing the symbol Bool,

• an infinite set U of sort parameters u,

• an infinite set X of variables x,

• an infinite set F of function symbols f containing the symbols ≈, ∧, and ¬,

• an infinite set A of attribute names a,

• an infinite set V of attribute values v,

• the setW of Unicode character strings w,

• a two-element set B = {true, false} of Boolean values b,

• the set N of natural numbers n,

• an infinite set T N of theory names T,

• an infinite set L of logic names L.

5.1 The language of sorts

In many-sorted logics, terms are typed, or sorted. Each sort, which stands for a non-empty set
of elements, is denoted by a sort symbol. In SMT-LIB logic, the language of sorts is extended
from sort symbols to sort terms built with symbols from the set S above. Formally, we have
the following.

Definition 1 (Sorts). For all non-empty subsets S of S and all mappings ar : S → N, the set
Sort(S) of all sorts over S (with respect to ar) is defined inductively as follows:

1. every s ∈ S with ar(s) = 0 is a sort;

2. If s ∈ S and ar(s) = n > 0 and σ1, . . . , σn are sorts, then the term s σ1 · · · σn is a sort.

We say that s ∈ S has (or is of) arity n if ar(s) = n.

(Attributes) α ::= a | a = v

(Patterns) p ::= x | f x∗

(Terms) t ::= x | f t∗ | f σ t∗

| ∃ (x:σ)+ t | ∀ (x:σ)+ t | let (x = t)+ in t | match t with (p→ t)+

| t α+

Figure 5.2: Abstract syntax for unsorted terms

As an example of a sort, if Int and Real are sort symbols of arity 0, and List and Array are
sort symbols of respective arity 1 and 2, then the expression List (Array Int (List Real)) and
all of its subexpressions are sorts.

Function symbol declarations in theory declarations (defined later) use also parametric sorts.
These are defined similarly to sorts above except that they can be built also over a further set
U of sort parameters, used like sort symbols of arity 0. Similarly to the example above, if u1, u2

are elements of U , the expression List (Array u1 (List u2)) and all of its subexpressions are
parametric sorts.

An abstract syntax for sorts σ and parametric sorts τ, which ignores arity constraints
for simplicity, is provided in Figure 5.1. Note that every sort is a parametric sort, but not
vice versa. Also note that parametric sorts are used only in theory declarations; they are
not part of the SMT-LIB logic proper. In the following, we say “sort” to refer exclusively to
non-parametric sorts.

5.2 The language of terms

In the abstract syntax, terms are built out of variables from X , function symbols from F , and
a set of binders. The logic considers, in fact, only well-sorted terms, a subset of all possible terms
determined by a sorted signature, as described below.

The set of all terms is defined by the abstract syntax rules of Figure 5.2. The rules do not
distinguish between constant and function symbols (they are all members of the set F ). These
distinctions are really a matter of arity, which is taken care of later by the well-sortedness
rules.

Binders

For all n ≥ 0, distinct variables x1, . . . , xn ∈ X , and sorts σ1, . . . , σn,

• the prefix construct ∃ x1:σ1 · · · xn:σn _ is a sorted existential binder (or existential quantifier)
for x1, . . . , xn;

• the prefix construct ∀ x1:σ1 · · · xn:σn _ is a sorted universal binder (or universal quantifier)
for x1, . . . , xn;

• the mixfix construct let x1 = _ · · · xn = _ in _ is a (parallel-)let binder for x1, . . . , xn.

• the mixfix construct match _ with p1 → _ · · · pn → _ is a match binder for the variables
that occur in the pattern pi for each i = 1, . . . , n.

Occurrences of variables in terms are defined to be free or bound as in the case of the
concrete syntax; the scope of each bound variable is defined similarly as well (see Subsec-
tion 3.6.3). Terms are closed if they contain no free variables, and open otherwise. Terms are
ground if they are variable-free.

For simplicity, the defined language does not contain any logical symbols other than the
binders. Logical connectives for negation, conjunction and so on and the equality symbol,
which we denote here by ≈, are just function symbols of the basic theory Core, implicitly
included in all SMT-LIB theories (see Subsection 3.7.1).

Annotations

Terms can be optionally annotated with zero or more attributes. Attributes have no logical
meaning, but they are a convenient mechanism for adding meta-logical information, as illus-
trated in Section 3.6. Syntactically, an attribute is either an attribute name a ∈ A or a pair the
form a = v where a ∈ A and v is an attribute value in V .1

Function symbols themselves may be annotated with a sort, as in f σ. Sort annotations
simplify the sorting rules of the logic, which determine the set of well-sorted terms.

5.2.1 Signatures

Well-sorted terms in SMT-LIB logic are terms that can be associated with a unique sort by
means of a set of sorting rules similar to typing rules in programming languages. The rules
are based on the following definition of a (many-sorted) signature.

Definition 2 (SMT-LIB Signature). An SMT-LIB signature, or simply a signature, is a tuple Σ
consisting of:

• a set ΣS ⊆ S of sort symbols containing Bool;

• a set ΣF ⊆ F of function symbols;

• a distinguished finite set ΣC ⊆ ΣF of constructor symbols;

• a distinguished finite set ΣG ⊆ ΣF of selector symbols, disjoint with ΣC;

• a distinguished finite set ΣT ⊆ ΣF of tester symbols, disjoint with ΣC and ΣG, and with
the same cardinality as ΣC;

• a total mapping conΣ : ΣS → 2ΣC
, assigning a (possibly empty) set of constructors to

each sort symbol;

• a total mapping ar : ΣS →N, assigning an arity to each sort symbol, with ar(Bool) = 0;

1At this abstract level, the syntax of attribute values is intentionally left unspecified.



5.2. THE LANGUAGE OF TERMS 73

• a total mapping selΣ : ΣC → (ΣG)∗, assigning a sequence of n distinct selectors to each
constructor of arity n so that no selector is assigned to more than one constructor;

• a bijective mapping tesΣ : ΣC → ΣT, assigning a tester to each constructor;

• a partial mapping from X to Sort(ΣS), assigning a sort to some of the variables in X ;2

• a left-total ranking relation3 R from ΣF to Sort(ΣS)+, assigning at least one rank to each
function symbol and such that

1. for all constructors c ∈ ΣC, selectors g1 · · · gn = selΣ(c), and sorts σ1, . . . , σn, σ ∈
Sort(ΣS), if (c, σ1 · · · σnσ) ∈ R then (gi, σσi) ∈ R for all i = 1, . . . , n;

2. for all constructors c ∈ ΣC and testers p = tesΣ(c), if (c, σ1 · · · σnσ) ∈ R then
(p, σBool) ∈ R;

3. there is no constructor c ∈ ΣC such that (c, σ̄1σ), (c, σ̄2σ) ∈ R for distinct σ̄1 and
σ̄2.(57)

A sort in Σ is an algebraic data type if it is assigned a non-empty set of constructors.

Remark 8. The restrictions imposed on theory declarations and on the various commands
for declaring new symbols in SMT-LIB scripts make sure that the signature defined by an
SMT-LIB script is in fact a signature in the sense of Definition 2.

In the following, we will write Sort(Σ) as an abbreviation of Sort(ΣS). We will work with
ranked function symbols and sorted variables in a signature. Formally, given a signature Σ, a
ranked function symbol is a pair ( f , σ1 · · · σnσ) in F ×Sort(Σ)+, which we write as f :σ1 · · · σnσ. A
sorted variable is a pair (x, σ) in X × Sort(Σ), which we write as x:σ. We write f :σ1 · · · σnσ ∈ Σ
and x:σ ∈ Σ to denote that f has rank σ1 · · · σnσ in Σ and x has sort σ in Σ.

We also consider signatures that differ from a given signature Σ only the sort they assign
to variables. We will also consider signatures that conservatively expand a given signature Σ
with additional sort and function symbols or additional ranks for Σ’s function symbols.

Definition 3 (Signature variants and expansions). A signature Σ′ is a variant of a signature Σ
if it is identical to Σ possibly except for its mapping from variables to sorts.

A signature Ω is a expansion of a signature Σ if all of the following hold: ΣS ⊆ ΩS;
ΣF ⊆ ΩF; the sort symbols of Σ have the same arity in Σ and in Ω; every sort of Σ has the
same constructors in Ω that it has in Σ; every constructor of Σ has the same selectors and
testers in Ω that it has in Σ; for all x ∈ X and σ ∈ Sort(Σ), x:σ ∈ Σ iff x:σ ∈ Ω; for all f ∈ F
and σ̄ ∈ Sort(Σ)+, if f :σ̄ ∈ Σ then f :σ̄ ∈ Ω. In that case, Σ is a subsignature of Ω.

Overloading

The rank of a function symbol in a signature specifies, in order, the expected sort of the
symbol’s arguments and result. Note that it is possible for a function symbol to be overloaded in

2Note that Sort(ΣS), the set of all sorts over ΣS, is non-empty because at least one sort in ΣS, Bool, has arity 0.
3A binary relation R ⊆ X×Y is left-total if for each x ∈ X there is (at least) a y ∈ Y such that xRy.

a signature Σ by being associated to more than one rank in Σ. This form of ad-hoc polymorphism
is entirely unrestricted: a function symbol can have completely different ranks—even varying
in arity. For example, in a signature with sorts Int and Real (with the expected meaning), it
is possible for the minus symbol − to have all of the following ranks: Real Real (for unary
negation over the reals), Int Int (for unary negation over the integers), Real Real Real (for
binary subtraction over the reals), and Int Int Int (for binary subtraction over the integers).

Together with the mechanisms used to declare theories (described in Section 5.4) and
algebraic datatypes (described in Section 4.2.3), overloading also provides an approximate
form of parametric polymorphism by allowing one to declare function symbols with ranks all
having the same shape. For instance, it is possible to declare an array access symbol with rank
(Array σ1 σ2) σ1 σ2 for all sorts σ1, σ2 in a theory signature. Strictly speaking, this is still ad-hoc
polymorphism because SMT-LIB logic itself does not allow parametric sorts.4 However, it
provides most of the convenience of parametric polymorphism while remaining within the
confines of the standard semantics of many-sorted FOL.

A function symbol can be ambiguous in an SMT-LIB signature for having distinct ranks
of the form σ̄σ1 and σ̄σ2 where σ̄ is a sequence of sorts and σ1 and σ2 are different sorts.
Thanks to the requirement in Definition 2 that variables have exactly one sort in a signature,
in signatures with no ambiguous function symbols every term can have at most one sort. In
contrast, with an ambiguous symbol f whose different ranks are σ̄σ1, . . . , σ̄σn, a term of the
form f t̄, where the terms t̄ have sorts σ̄, can be given a unique sort only if f is annotated with
one of the result sorts σ1, . . . , σn, that is, only if it is written as f σi t̄ for some i ∈ {i, . . . , n}.

5.2.2 Well-sorted terms

Figure 5.3 provides a set of rules defining well-sorted terms with respect to an SMT-LIB sig-
nature Σ. Strictly speaking then, and similarly to more conventional logics, the SMT-LIB logic
language is a family of languages parametrized by the signature Σ. As explained later, for
each script working in the context of a background theory T , the specific signature is jointly
defined by the declaration of T plus any additional sort and function symbol declarations
contained in the script.

The format and meaning of the sorting rules in Figure 5.3 is fairly standard and should
be largely self-explanatory to readers familiar with type systems. In more detail, the letter σ

(possibly primed or with subscripts) denotes sorts in Sort(Σ), the letter k denotes a natural
number. Expressions of the form Σ[x1 : σ1, . . . , xn : σn] denotes the signature that maps xi to
sort σi for i = 1, . . . , n, and coincides otherwise with Σ. Finally, α∗ denotes a possibly empty
sequence of attributes. The rules operate over sorting judgments which are triples of the form
Σ ` t : σ.

Definition 4 (Well-sorted Terms). For every SMT-LIB signature Σ, a term t generated by the
grammar in Figure 5.2 is well-sorted (with respect to Σ) if Σ ` t : σ is derivable by the sorting
rules in Figure 5.3 for some sort σ ∈ Sort(Σ). In that case, we say that t has, or is of, sort σ.

With this definition, it is possible to show that every term has at most one sort.(58)

4Parametric sort terms that occur in theory declarations and algebraic datatype declarations are meta-level
syntax as far as SMT-LIB logic is concerned. They are schemas standing for concrete sorts.

Σ ` x α∗ : σ
if x:σ ∈ Σ

Σ ` t1 : σ1 · · · Σ ` tk : σk

Σ ` ( f t1 · · · tk) α∗ : σ
if

{
f :σ1 · · · σkσ ∈ Σ,

f :σ1 · · · σkσ′ /∈ Σ for all σ′ 6= σ

Σ ` t1 : σ1 · · · Σ ` tk : σk

Σ ` ( f σ t1 · · · tk) α∗ : σ
if

{
f :σ1 · · · σkσ ∈ Σ,

f :σ1 · · · σkσ′ ∈ Σ for some σ′ 6= σ

Σ[x1:σ1, . . . , xk+1:σk+1] ` t : Bool

Σ ` (Qx1:σ1 · · · xk+1:σk+1 t) α∗ : Bool
if Q ∈ {∃, ∀}

Σ ` t1 : σ1 · · · Σ ` tk+1 : σk+1

Σ[x1:σ1, . . . , xk+1:σk+1] ` t : σ

Σ ` (let x1 = t1 · · · xk+1 = tk+1 in t) α∗ : σ

if x1, . . . , xk+1 are all distinct

Σ ` t : δ

Σ[x̄i:σ̄i] ` ti : σ for i = 1, . . . , k + 1

Σ ` (match t with c1 x̄1 → t1 · · · ck+1 x̄k+1 → tk+1) α∗ : σ

if


{c1, . . . , ck+1} = conΣ(δ),

for all i = 1, . . . , k + 1

ci:σ̄iδ ∈ Σ and

x̄i contains no repetitions

Σ ` t : δ

Σ[x̄i:σ̄i] ` ti : σ for i = 1, . . . , k
Σ[xi:δ] ` ti : σ for i = k + 1, . . . , n

Σ ` (match t with p1 → t1 · · · pn → tn) α∗ : σ

if



{c1, . . . , ck} ⊆ conΣ(δ) 6= ∅,

{p1, . . . , pn} = {c1 x̄1, . . . , ck x̄k} ∪
{xk+1, . . . , xn},

k < n,

for all i = 1, . . . , k
ci:σ̄iδ ∈ Σ and

x̄i contains no repetitions

Figure 5.3: Well-sortedness rules for terms.

Remark 9 (Match rules). The two rules for the match binder in Figure 5.3 require that the
match cases be exhaustive: every constructor term of sort δ must match one of the patterns; but
allow it to be redundant: the same term may match more than one pattern. Exhaustiveness is
necessary to make sure each match expression is semantically well defined. The first rule deals
with match expressions where no patterns consist of a variable. In that case, exhaustiveness is
enforced by requiring that each constructor of the datatype appear in one of the patterns. The
second rule deals with match expressions where one or more patterns consist of a variable.
In that case, exhaustiveness is guaranteed simply by the presence of those variables. In both
cases, the preconditions ensure that δ is a datatype, not just any sort, by requiring it to have a
non-empty set of constructors.

Definition 5 (SMT-LIB formulas). For each signature Σ, the language of SMT-LIB logic is the
set of all well-sorted terms wrt Σ. Formulas are well-sorted terms of sort Bool.

In the following, we will use ϕ and ψ to denote formulas.

Constraint 3. SMT-LIB scripts consider only closed formulas, or sentences, closed terms of sort
Bool.(59)

There is no loss of generality in the restriction above because, as far as satisfiability is
concerned, every formula ϕ with free variables x1, . . . , xn of respective sort σ1, . . . , σn, can be
rewritten as

∃ x1:σ1 . . . xn:σn ϕ .

An alternative way to avoid free variables in scripts is to replace them by fresh constant
symbols of the same sort. This is again with no loss of generality because, for satisfiability
modulo theories purposes, a formula’s free variables can be treated equivalently as free symbols
(see later for a definition).

5.3 Structures and Satisfiability

The semantics of SMT-LIB is essentially the same as that of conventional many-sorted logic,
relying on a similar notion of Σ-structure.

Definition 6 (Σ-structure). Let Σ be a signature. A Σ-structure A is a pair consisting of a set
A, the universe of A, and a mapping that interprets

• each σ ∈ Sort(Σ) as a non-empty subset σA of A, which we call the domain of σ in A, with
A =

⋃
σ∈ΣS σA;(60)

• each (ranked function symbol) f :σ ∈ Σ as an element ( f :σ)A of σA;

• each f :σ1 · · · σnσ ∈ Σ with n > 0 as a total function ( f :σ1 · · · σnσ)A from σA
1 × · · · × σA

n
to σA.

If B is an Ω-signature with universe B and Σ is a subsignature of Ω, the reduct of B to Σ is
the (unique) Σ-structure with universe B that interprets its sort and function symbols exactly
as B. A structure B is an expansion of a Σ-structure A if A is the Σ-reduct of B.

Note that, as a consequence of overloading, a Σ-structure does not interpret plain function
symbols but ranked function symbols. Also note that any Σ-structure is also a Σ′-structure for
every variant Σ′ of Σ.

Definition 7 (Absolutely free structure). Let A be a Σ-structure with universe A and let G ⊆ A.
Let ΣG be the expansion of Σ obtained by adding to Σ a constant symbol a of rank σ for every
a ∈ G and sort σ ∈ ΣS such that a ∈ σA. Then, A is an absolutely free structure (with generators
G) if

• for all σ ∈ ΣS, σA is the set of well-sorted ground terms of signature ΣG;

• A interprets every function symbol f :σ1 · · · σnσ ∈ ΣG as the function that maps each
tuple (t1, . . . , tn) ∈ σA

1 × . . .× σA
n to the term f (t1, . . . , tn).

Intuitively, an absolutely free Σ-structure with a set G of generators interprets every well-
sorted ground ΣG-term as itself. Note that the choice of generators affects the property of
being absolutely free. For instance, no structure without constant symbols can be absolutely
free with an empty set of generators.

The SMT-LIB logic considers only structures that interpret in a special way the sort Bool
and any constructor, selector, and tester symbols in their signature.

Definition 8 (SMT-LIB Σ-structure). An SMT-LIB structure is a Σ-structure A such that

1. BoolA = B = {false, true} with false and true distinct;

2. if Ω is the signature obtained from Σ by removing all of its non-constructor function
symbols, the Ω-reduct of A is an absolutely free algebra with generators

⋃
σ∈S σA where

S collects the sorts of Σ that are not datatypes;

3. for all constructors c:σ1 · · · σnσ ∈ Σ with n > 0, selectors g1:σσ1, . . . , gn:σσn with selΣ(c) =
g1 · · · gn, values (v1, . . . , vn) ∈ σA

1 × · · · × σA
n , and i = 1, . . . , n,

gA
i (c

A(v1, . . . , vn)) = vi ;

4. for all constructors c:σ̄σ ∈ Σ, testers q with tesΣ(c) = q, and values v ∈ σA,

qA(v) = true iff v is in the range of cA.

From now on, when we say “structure” we will mean “SMT-LIB structure.”

Remark 10. The restrictions in SMT-LIB structures on the interpretation of constructors, se-
lectors and testers, together with the well-foundedness restrictions on those constructs in
signatures, guarantee that sorts with constructors indeed denote algebraic datatypes as tradi-
tionally understood in the literature.

Remark 11 (Partiality of selectors). As in classical first-order logic, all function symbols in a
signature Σ are interpreted as total functions in a Σ-structure A. This means in particular that
if g:σσi ∈ Σ is a selector, the function gA returns a value even for inputs outside the range
of g’s constructor. Definition 8 imposes no constraints on that value, other than it belongs to
σA

i . For instance, in a structure A with a sort for integer lists with constructors nil and insert
and selectors head and tail for insert, the function headA maps nilA to some integer value.
Similarly, tailA maps nilA to some integer list. This is consistent with the general modeling
of partial functions in SMT-LIB as underspecified total functions—which requires a solver to
consider all possible (well-sorted) ways to make a partial function total.

The notion of isomorphism between structures introduced below is needed for Defini-
tion 12, Theory Combination, in Section 5.4.

Definition 9 (Isomorphism). Let A and B be two Σ-structures with respective universes A
and B. A mapping h : A→ B is a homomorphism from A to B if

1. for all σ ∈ Sort(Σ) and a ∈ σA,
h(a) ∈ σB ;

2. for all f :σ1 . . . σnσ ∈ Σ with n > 0 and (a1, . . . , an) ∈ σA
1 × · · · × σA

n ,

h(( f :σ1 . . . σnσ)A(a1, . . . , an)) = ( f :σ1 . . . σnσ)B(h(a1), . . . , h(an)) .

A homomorphism between A and B is an isomorphism of A onto B if it is invertible and its
inverse is a homomorphism from B to A.

Two Σ-structures A and B are isomorphic if there is an isomorphism from one onto the other.
Isomorphic structures are interchangeable for satisfiability purposes because one satisfies a set
of Σ-sentences if and only if the other one does.

5.3.1 The meaning of terms

A valuation into a Σ-structure A is a partial mapping v from X × Sort(Σ) to the set of all
domain elements of A such that, for all x ∈ X and σ ∈ Sort(Σ), v(x:σ) ∈ σA. We denote by
v[x1:σ1 7→ a1, . . . , xn:σn 7→ an] the valuation that maps xi:σi to ai ∈ σA

i for i = 1, . . . , n and is
otherwise identical to v.

If v is a valuation into Σ-structure A, the pair I = (A, v) is a Σ-interpretation. We write
I [x1:σ1 7→ a1, . . . , xn:σn 7→ an] as an abbreviation for the Σ′-interpretation

(A′, v[x1:σ1 7→ a1, . . . , xn:σn 7→ an])

where Σ′ = Σ[x1:σ1, . . . , xn:σn] and A′ is just A but seen as a Σ′-structure.
A Σ-interpretation I assigns a meaning to well-sorted Σ-terms by means of a uniquely

determined (total) mapping [[_]]I of such terms into the universe of its structure.

Definition 10. Let Σ be an SMT-LIB signature and let I be a Σ-interpretation. For every
well-sorted term t of sort σ with respect to Σ, [[t]]I is defined recursively as follows.

1. [[u α1 · · · αn]]I = [[u]]I

2. [[x]]I = v(x:σ) if I = (A, v)

3. [[ f̂ t1 . . . tn]]I = ( f :σ1 · · · σnσ)A(a1, . . . , an) if


I = (A, v) with signature Σ,

f̂ = f or f̂ = f σ,

for i = 1, . . . , n
Σ ` ti : σi and ai = [[ti]]

I

4. [[let x1 = t1 · · · xn = tn in t]]I = [[t]]I
′

if


I has signature Σ,

for i = 1, . . . , n
Σ ` ti : σi and ai = [[ti]]

I ,

I ′ = I [x1:σ1 7→ a1, . . . , xn:σn 7→ an]

5. [[∃ x1:σ1 · · · xn:σn t]]I = true iff [[t]]I
′
= true for some I ′

such that


I ′ = I [x1:σ1 7→ a1, . . . , xn:σn 7→ an],

(a1, . . . , an) ∈ σA
1 × · · · × σA

n ,

I = (A, v),

6. [[∀ x1:σ1 · · · xn:σn t]]I = true iff [[t]]I
′
= true for all I ′

such that


I ′ = I [x1:σ1 7→ a1, . . . , xn:σn 7→ an],

(a1, . . . , an) ∈ σA
1 × · · · × σA

n ,

I = (A, v),

7. [[match t with c→ t0 p1 → t1 · · · pn → tn]]I = [[t0]]I

if

{
I = (A, v),
[[t]]I is in the range of cA

8. [[match t with (c x1 · · · xk+1)→ t0 p1 → t1 · · · pn → tn]]I =

[[let x1 = g1(t) · · · xk+1 = gk+1(t) in t0]]I

if


I = (A, v) with signature Σ,

[[t]]I is in the range of cA,

selΣ(c) = g1 · · · gk+1

9. [[match t with (c x̄)→ t0 p1 → t1 · · · pn → tn]]I =

[[match t with p1 → t1 · · · pn → tn (c x̄)→ t0]]I

if

{
I = (A, v),
[[t]]I is not in the range of cA

10. [[match t with x0 → t0 p1 → t1 · · · pn → tn]]I = [[let x0 = t in t0]]I

One can show that [[_]]I is well-defined, and hence total, over closed terms that are well-sorted
with respect to I ’s signature.

A Σ-interpretation I satisfies a Σ-formula ϕ if [[ϕ]]I = true, and falsifies it if [[ϕ]]I = false.
The formula ϕ is satisfiable if there is a Σ-interpretation I that satisfies it, and is unsatisfiable
otherwise.

For a closed term t, its meaning [[t]]I in an interpretation I = (A, v) is independent of the
choice of the valuation v—because the term has no free variables. For such terms then, we
can write [[t]]A instead of [[t]]I . Similarly, for sentences, we can speak directly of a structure
satisfying or falsifying the sentence. A Σ-structure that satisfies a sentence is also called a
model of the sentence.

5.4 Theories

Theories are traditionally defined as sets of sentences. Alternatively, and more generally, in
SMT-LIB a theory is defined as a class of structures with the same signature.

Definition 11 (Theory). For any signature Σ, a Σ-theory is a class of Σ-structures. Each of these
structures is a model of the theory.

Typical SMT-LIB theories consist of a single model (e.g., the integers) or of the class of all
structures that satisfy some set of sentences—the axioms of the theory. Note that in SMT-LIB
there is no requirement that the axiom set be finite or even recursive.

5.4.1 Combined Theories

SMT-LIB uses both basic theories, obtained as instances of a theory declaration schema, and
combined theories, obtained by combining together suitable instances of different theory sche-
mas. The combination mechanism is defined below.

Two signatures Σ1 and Σ2 are compatible if they have the same sort symbols, have the same
datatype constructors, selectors and testers, and they agree on both the arity they assign to
sort symbols and the sorts they assign to variables.5 Two theories are compatible if they have
compatible signatures. The combination Σ1 + Σ2 of two compatible signatures Σ1 and Σ2 is the
smallest compatible signature that is an expansion of both Σ1 and Σ2, i.e., the unique signature
Σ compatible with Σ1 and Σ2 such that, for all f ∈ F and σ̄ ∈ Sort(Σ)+, f :σ̄ ∈ Σ iff f :σ̄ ∈ Σ1

or f :σ̄ ∈ Σ2.

Definition 12 (Theory Combination). Let T1 and T2 be two theories with compatible signatures
Σ1 and Σ2, respectively. The combination T1 +T2 of T1 and T2 consists of all (Σ1 +Σ2)-structures
whose reduct to Σi is isomorphic to a model of Ti, for i = 1, 2.

Over pairwise compatible signatures the signature combination operation + is associative
and commutative. The same is also true for the theory combination operation + over com-
patible theories. This induces, for every n > 1, a unique n-ary combination T1 + · · · + Tn

of mutually compatible theories T1, . . . , Tn in terms of nested binary combinations. Combined

5Observe that compatibility is an equivalence relation on signatures.

(Sort symbol declarations) sdec ::= s n α∗

(Fun. symbol declarations) fdec ::= f σ+ α∗

(Param. fun. symbol declarations) pdec ::= fdec | Π u+ ( f τ+ α∗)

(Theory attributes) tattr ::= sorts = sdec+ | funs = pdec+

| sorts-description = w
| funs-description = w
| definition = w | values = t+

| notes = w | α

(Theory declarations) tdec ::= theory T tattr+

Figure 5.4: Abstract syntax for theory declarations

theories in SMT-LIB are exclusively theories of the form T1 + · · ·+ Tn for some basic SMT-LIB
theories T1, . . . , Tn.

SMT is about checking the satisfiability or the entailment of formulas modulo some (pos-
sibly combined) theory T . This standard adopts the following precise formulation of such
notions.

Definition 13 (Satisfiability and Entailment Modulo a Theory). For any Σ-theory T , a Σ-
sentence is satisfiable in T iff it is satisfied by one of T ’s models. A set Γ of Σ-sentences
T -entails a Σ-sentence ϕ, written Γ |=T ϕ, iff every model of T that satisfies all sentences in Γ
satisfies ϕ as well.

5.4.2 Theory declarations

In SMT-LIB, basic theories are obtained as instances of theory declarations. (In contrast, com-
bined theories are defined in logic declarations.) An abstract syntax of theory declarations
is defined in Figure 5.4. The symbol Π in parametric function symbol declarations is a (uni-
versal) binder for the sort parameters—and corresponds to the symbol par in the concrete
syntax.

To simplify the meta-notation let T denote a theory declaration with theory name T. Given
such a theory declaration, assume first that T has no sorts-description and funs-description
attributes, and let S and F be respectively the set of all sort symbols and all function symbols
occurring in T. Let Ω be a signature whose set of sort symbols that are not datatypes includes all
the symbols in S, with the same arity. The definition provided in the definition attribute of T
must be such that every signature like Ω above uniquely determines a theory T̂ = T[Ω] as an
instance of T with signature Ω̂ defined as follows:

1. Ω̂S = ΩS and Ω̂F = F ∪ΩF,

2. no variables are sorted in Ω̂,(61)

3. for all f ∈ Ω̂F and σ̄ ∈ (Ω̂S)+,

(Logic attributes) lattr ::= theories = T+ | language = w
| extensions = w | values = w
| notes = w | α

(Logic declarations) ldec ::= logic L lattr+

Figure 5.5: Abstract syntax for logic declarations

f :σ̄ ∈ Ω̂ iff

(a) f :σ̄ ∈ Ω, or

(b) T contains a declaration of the form f σ̄ ᾱ, or

(c) T contains a declaration of the form Π ū ( f τ̄ ᾱ) and σ̄ is an instance of τ̄.

4. A function symbol of Ω̂ is a datatype constructor/selector/tester in Ω̂ iff it is so in Ω.

We say that a ranked function symbol f :σ̄ of Ω̂ is declared in T if f :σ̄ ∈ Ω̂ because of
Point 3b or 3c above. The free sort symbols of T̂ are the sort symbols of Ω̂ that are not in S
and are not datatypes. Similarly, the free function symbols of T̂ are the ranked function symbols
of Ω̂ that are not declared in T and are not datatype constructors, selectors or testers.6 This
terminology is justified by the following additional requirement on T.

The definition of T must be parametric, in this sense: it must not constrain the free symbols
of any instance T[Ω] of T in any way. Technically, T must be defined so that the set of models
of T[Ω] is closed under any changes in the interpretation of the free symbols. That is, every
structure obtained from a model of T[Ω] by changing only the interpretation of T[Ω]’s free
symbols should be a model of T[Ω] as well.(62)

The case of theory declarations with sorts-description and funs-description attributes is
similar.

5.5 Logics

A logic in SMT-LIB is any sublogic of the main SMT-LIB logic obtained by

• fixing a signature Σ and a Σ-theory T ,

• restricting the set of structures to the models of T , and

• restricting the set of sentences to some subset of the set of all Σ-sentences.

A model of a logic with theory T is any model of T ; a sentence is satisfiable in the logic iff
it is satisfiable in T .

6Note that because of overloading we talk about ranked function symbols being free or not, not just function
symbols.

5.5.1 Logic declarations

Logics are specified by means of logic declarations. Contrary to the theory declarations, a
logic declaration specifies a single logic, not a class of them, so we call the logic L too. An
abstract syntax of logic declarations is defined in Figure 5.5.

Let L be a logic declaration whose theories attribute has value T1, . . . , Tn.

Theory. The logic’s theory is the theory T uniquely determined as follows. For each i =
1, . . . , n, let Si be the set of all sort symbols occurring in Ti. The text in the language attribute
of L may specify an additional set S0 of sort symbols and an additional set of ranked function
symbols with ranks over Sort(S)+ where S =

⋃
i=0,...,n Si. Let Ω be the smallest signature with

ΩS = S containing all those ranked function symbols. Then for each i = 1, . . . , n, let Ti[Ω] be
the instance of Ti determined by Ω as described in Subsection 5.4.2. The theory of L is

T = T1[Ω] + · · ·+ Tn[Ω] .

Note that T is well defined. To start, Ω is well defined because any sort symbols shared by
two declarations among T1, . . . , Tn have the same arity in them. The theories T1[Ω], . . . , Tn[Ω]
are well defined because Ω satisfies the requirements in Subsection 5.4.2. Finally, the signa-
tures of T1[Ω], . . . , Tn[Ω] are pairwise compatible because they all have the same sort symbols,
each with the same arity in all of them.

Values. The values attribute is expected to designate for each sort σ of the logic’s theory T
a distinguished set Vσ of ground terms called values. The definition of Vσ should be such that
every sentence satisfiable in the logic L is satisfiable in a model A of T where each element of
σA is denoted by some element of Vσ. In other words, if Σ is T ’s signature, A is such that, for
all σ ∈ ΣS and all a ∈ σA, a = [[t]]A for some t ∈ Vσ. For example, in a logic of the integers,
the set of values for the integer sort might consist of all the terms of the form 0 or [−]n where
n is a non-zero numeral.

For flexibility, we do not require that Vσ be minimal. That is, it is possible for two terms
of Vσ to denote the same element of σA. For example, in a logic of rational numbers, the set
of values for the rational sort might consist of all the terms of the form [−]m/n where m is a
numeral and n is a non-zero numeral. This set covers all the rationals but, in contrast with the
previous example, is not minimal because, for instance, 3/2 and 9/6 denote the same rational.

Note that the requirements on Vσ can be always trivially satisfied by L by making sure that
the signature Ω above contains a distinguished set of infinitely many additional free constant
symbols of sort σ, and defining Vσ to be that set. We call these constant symbols abstract values.
Abstract values are useful to denote the elements of uninterpreted sorts or sorts standing for
structured data types such as lists, arrays, sets and so on.7

Recall that algebraic datatypes are not defined in theories but directly at the level of the
underlying SMT-LIB logic. For each such sort, the set of values is fixed to the set of constructor
terms built over values from other sorts. For example, in a parametric list datatype δ with the

7The concrete syntax reserves a special format for constant symbols used as abstract values: they are members
of the 〈symbol〉 category that start with the character @.

usual nil and cons constructors, the set of values for δ consists of nil and all terms of the form
cons v1 (cons v2 · · · (cons vn nil) · · · ) where n > 0 and v1, . . . , vn are (recursively) values of
the same sort.

APPENDIX A

Notes

1 To define such theory signatures formally, SMT-LIB would need to rely on a more powerful underlying logic,
for instance one with dependent types.

2 Preferring ease of parsing over human readability is reasonable in this context not only because SMT-LIB
benchmarks are meant to be read by solvers but also because they are produced in the first place by automated
tools like verification condition generators or translators from other formats.

3 The move to the Unicode character standard was motivated by the fact that US-ASCII is is inadequate in
international settings and Unicode has become the dominant standard.

4 This syntactical category excludes the non-English letters of Unicode because it is used to define identifiers,
which traditionally use only English letters. Future versions might extend it to non-English letters as well.

5 This is to achieve maximum generality and independence from programming language conventions. This
way, SMT-LIB theories of strings that use string literals as constant symbols have the choice to define certain
string constants, such as "\n" and "\012", as equivalent or not. If we used, say, C-style backslash-prefixed
escape sequences at the SMT-LIB level, it would be impractical and possibly confusing to represent literally
certain sequences of characters. For instance, with C-style conventions the literals "\\e" and "\e" would be
parsed as the same two-character literal consisting of the characters 5Chex and 65hex. To represent the three-
character string literal consisting of the characters 5Chex5Chex65hex, one would have to write, for instance,
something like "\\\e" or "\x5C\e".

6 Strictly speaking, command names do not need to be reserved words because of the language’s namespace
conventions. Having them as reserved words, however, simplifies the creation of compliant parsers with the
aid of parser generators such as Lex/YACC and ANTLR.

7 Backslash is disallowed in quoted symbols just for simplicity. Otherwise, for Common Lisp compatibility
they would have to be treated as an escaping symbol (see Section 2.3 of [Ste90]).

8 Symbols have been added as indices in Version 2.6 for increased flexibility.

9 The sorts of pattern variables are not specified explicitly since they can be readily inferred from the datatype.

10 Such lists are semantically redundant in the sense that some of the cases can be dropped without affecting its
meaning. This is allowed to simplify parsing.

11 This guarantees that the match expressions are always well defined (see Section 5.3).

12 The restriction to flat patterns is for simplicity and may be lifted in later versions.

13 This shadowing rule follows common lexical scoping conventions. It is easy to avoid declarations like (par
(Int) (f (Array Int Int) Int)), which are potentially confusing to human readers, simply by a more
judicious choice of parameter names. In patterns like (cons x nil) are problematic

14 The reason for this exception is mostly historical. Many SMT solvers apply this restriction, possibly under
the assumption that shadowing a theory symbol is, more often than not, unintended and so best treated as an
error.

15 This restriction eliminates the need for a solver to do sort inference in order to determine the sorts of terms
containing applications of ambiguous function symbols. The use of as is not required in patterns because
there sort inference is needed anyway since pattern variables are not explicitly assigned a sort.

16 The reason patterns annotate the body of a formula they refer to as opposed to the formula itself, is that, this
way, they can use the body’s free variables as pattern variables.

17 The rationale for allowing user-defined attributes is the same as in other attribute-value-based languages
(such as, e.g., BibTeX). It makes the SMT-LIB format more flexible and customizable. The understanding is
that user-defined attributes are allowed but need not be supported by an SMT solver for the solver to be
considered SMT-LIB compliant. We expect, however, that with continued use of the SMT-LIB format, certain
user-defined attributes will become widely used. Those attributes might then be officially adopted into the
format (as non-user-defined attributes) in later versions.

18 See the point made in Note 19.

19 Ideally, it would be better if :definition were a formal attribute, to avoid ambiguities and misinterpretation
and possibly allow automatic processing. The choice to use free text for this attribute is for practical reasons.
The enormous amount of effort that would be needed to first devise a formal language for this attribute
and then specify its value for each theory in the library is not justified by the current goals of SMT-LIB.
Furthermore, this attribute is meant mainly for human readers, not programs, hence a mathematically rigorous
definition in natural language seems enough.

20 Version 1.2 allowed one to specify a finitely-axiomatizable theory formally by listing a set of axioms in an
:axioms attribute. This attribute is gone from Version 2.0 onwards, because only one or two theories in the
SMT-LIB catalog can be defined that way. The remaining ones require infinitely many axioms or axioms with
quantified sort symbols which are not expressible in the language.

21 One advantage of defining instances of theory declaration schemas this way is that with one instantiation
of the schema one gets a single theory with arbitrarily nested sorts—another example being the theory of all
nested lists of integers, say, with sorts (List Int), (List (List Int)), etc. This is convenient in applications
coming from software verification, where verification conditions can contain arbitrarily nested data types. But
it is also crucial in providing a simple and powerful mechanism for theory combination, as explained later.

22 The reason for informal attributes is similar to that for theory declarations.

23 The attribute is text-valued because it is mostly for documentation purposes for the benefit of benchmark
users. A natural language description of the logic’s language seems therefore adequate for this purpose. Of
course, it is also possible to specify the language at least partially in some formal fashion in this attribute, for
instance by using BNF rules.

24 This is useful because in common practice, the syntax of a logic is often extended for convenience with
syntactic sugar.

25 This enables applications reading a compliant solver’s response to know when an identifier (like success) has
been completely printed and, in general, when the solver has completed processing a command. For example,
this is needed if one wants to use an off-the-shelf S-expression parser (e.g., read in Common Lisp) to read
responses.

26 The motivation for allowing these two behaviors is that the first one (exiting immediately when an error
occurs) may be simpler to implement, while the latter may be more useful for applications, though it might
be more burdensome to support the semantics of leaving the state unmodified by the erroneous command.

27 It is desirable to have the ability to remove declarations and definitions, for example if they are no longer
needed at some point during an interaction with a solver (so that the memory required for them can be re-
claimed), or if a defined symbol needs to be redefined. The current approach of allowing declarations and
definitions to be locally scoped supports removal by popping the containing assertion level. Other approaches,

such as the ability to add shadowing declarations or definitions of symbols, or to “undefine” or “undeclare”
them, present some issues: for example, how to print symbols that have been shadowed, undefined or unde-
clared. As a consequence, they are not supported in the language.

28 Setting :global-declarations to true can be understood as stating that declarations and definitions are not
part of the assertion stack, and so resetting the stack has no impact on them. This option is convenient in
applications that benefit from using push and pop for assertions but need to continue using symbols declared
after a push even after the corresponding pop.

29 The motivation for not overloading user-defined symbols is to simplify their processing by a solver. This
restriction is significant only for users who want to extend the signature of the theory used by a script with
a new polymorphic function symbol—i.e., one whose rank would contain parametric sorts if it was a theory
symbol. For instance, users who want to declare a “reverse” function on arbitrary lists, must define a different
reverse function symbol for each (concrete) list sort used in the script. This restriction might be removed in
future versions.

30 Note that this option is not intended to be used for comparisons between different solvers since they can
implement it differently. Its purpose is simply to guarantee the reproducibility of an individual solver’s
results under the same external conditions.

31 This is to avoid confusion with the responses to commands, which are written to the regular output channel.

32 Some commonly used statistics (e.g., number of restarts of a solver’s propositional reasoning engine) are
difficult to define precisely and generally, while the exact semantics of others, such as time and memory
usage, have not being agreed upon yet by the SMT community.

33 This command is useful for interactive use, to keep track of the current number of nested push commands.

34 This allows the user to reset the state of the solver without paying the cost of restarting it.

35 Having ALL is convenient for client applications that generate problems on the fly in a variety of logics
supported by some specific solver without knowing in advance the specific logic for each problem. SMT-LIB
scripts meant for the SMT-LIB library should not use ALL in set-logic but should instead use a specific logic
name.

36 The rationale is that a solver may need to make substantial changes to its internal configuration to provide the
functionality requested by these options, and so it needs to be notified before it starts processing assertions.

37 Strictly speaking, only sort symbols introduced with declare-sort expand the initial signature of theory sort
symbols. Sort symbols introduced with define-sort do not. They do not construct real sorts, but aliases of
sorts built with theory sort symbols and previously declared sort symbols.

38 The input of this command is split in two arguments precisely to facilitate the type-checking of the sort terms
τ1, . . . , τm. The first argument makes sure that when it is time to parse these terms each symbol δi is known to
be a sort symbol of arity ki.

39 The various restrictions on the definition of a datatype δ are crucial since they allow the existence of standard
interpretations for δ. See Chapter 5 for more details.

40 The rationale for this restriction is the same as for function symbols introduced with declare-fun and
define-fun, i.e., keep parsing and type checking simple.

41 Similar to define-fun, while strictly not needed, define-funs-rec provides a more structured way to define
functions axiomatically, as opposed to introducing a new function with declare-fun and then providing its
definition with assert and a quantified formula. This gives an SMT solver the opportunity to easily recognize
function definitions and possibly process them specially.

42 The motivation is to enable interactive users to see easily (exactly) which assertions they have asserted,
without having to keep track of that information themselves.

43 The motivation for having this command is that it corresponds to a common usage pattern with SMT solvers
which can be implemented considerably more efficiently than the general stack-based mechanism needed to
support push and pop commands. The restriction of assumptions to Boolean constants or negated constants
only is also motivated by efficiency concerns since user-defined constants are associated with formulas that
have already been internally simplified at definition time.

44 SMT solvers are incomplete for certain logics, typically those that include quantified formulas. However,
even when they are unable to determine whether the set Γ of all assertions and assumptions is satisfiable or
not, SMT solvers can typically compute a model for a set Γ′ of formulas that is entailed by Γ in the logic.
Interpretations in this model are often useful to client applications even if they are not guaranteed to come
from a model of Γ.

45 Abstract values are useful for reporting model values in logics containing for example the theory of arrays (see
Figure 3.4). For instance, a solver may specify the content of an array a of sort (Array Int Int) at positions
0–2 by returning the expression

(define-fun a () (store (store (store (as @array1 (Array Int Int)) 0 0) 1 2) 2 4)) .

The elements of a with index outside the 0–2 range are left unspecified because the array @array1 itself is left
unspecified.

46 Since it focuses only on preselected, Boolean terms, get-assignment can be implemented much more effi-
ciently than the more general get-value.

47 The rationale for providing the interpretation of a function symbol as a define command is that (i) the
syntax of such commands is general enough to be able to express a large class of functions symbolically in the
language of the current logic (possibly augmented with abstract values), and (ii) in principle the user could use
the provided definition as is in later interactions with the solver—as long as the original symbol’s declaration
is no longer in the current context as a consequence of a system restart, or a reset or a pop operation.

48 The current requirements on the returned definitions are rather weak. For instance, they allow a solver to
return something like

(define-fun-rec f ((x1 σ1) · · · (xn σn)) σ (f x1 · · · xn))

for a given f of rank σ1 · · · σnσ. Similarly, for constant symbols c of a sort σ that admits abstract values, they
allow a solver to return (define-fun c () @a) with @a abstract.
The reason for such weak requirements is that stronger ones are currently difficult to achieve in general be-
cause of limitations in the expressive power of some SMT-LIB theories/logics or in the computational abilities
of present SMT solvers. For instance, the current theory of arrays (see Figure 3.4) does not have enough con-
structor symbols to allow one to represent an array uniquely as a value term. As shown in a previous note,
one can use terms like

(store (store (store (as @array1 (Array Int Int)) 0 0) 1 2) 2 4))

which fixes only a portion of the array. This term has infinitely many interpretations that differ on the elements
at indices outside the 0–2 range. Similarly, because of the progress in automated synthesis, it is conceivable
that future solvers will be able to construct a model where a user-declared function symbol f denotes the
factorial function over the non-negative integers. In that case, a definition like

(define-fun-rec f ((x Int)) Int (ite (= x 0) 1 (* x (f (- x 1)))))

would not have a unique interpretation because it does not uniquely determine the behavior of f over the
negative integers. In contrast, the definition

(define-fun-rec f ((x Int)) Int (ite (<= x 1) 1 (* x (f (- x 1)))))

say, would determine a unique function over the whole set of integers.

49 The lax requirement is justified by the fact that the minimization problem alone is NP-hard in general. On the
other hand, it allows a solver to be compliant by just returning the same sequence given to (the most recent)
check-sat-assuming.

50 There is, as yet, no standard SMT-LIB proof format.

51 Unsatisfiable cores are useful for applications because they circumscribe the source of unsatisfiability in the
asserted set. The labeling mechanism allows users to track only selected asserted formulas when they already
know that the rest of the asserted formulas are jointly satisfiable.



93

52 Interjecting echo commands in a script can help a software client know where the solver is in the execution
of the script’s commands.

53 This is particularly useful for scripts that are used as benchmarks, as set-info can be used to store such
information as authors, date, expected response for a check command, difficulty level, and so on.

54 Note that the three possible values are strings and so need to be in quotes. The reasons for the values to be
strings as opposed to symbols is historical.

55 Having an explicit unknown value is useful for comparative evaluation of solvers, for example in the SMT-
COMP competition.

56 The reason for this unfortunate choice of names is historical. It is being kept only for backward compatibility
with previous versions.

57 Because of this constraint, the return sort of a constructor uniquely determines the sort of its arguments. That
removes the need to specify the sort of the pattern variables in a match expression.

58 It would have been reasonable to adopt an alternative version of the rule for well-sortedness of terms
( f σ t1 · · · tk) α∗ with annotated function symbols f σ, without the second conjunct of the rule’s side con-
dition. This would allow formation of terms with annotated function symbols f σ, even when f lacked two
ranks of the forms σ1 · · · σkσ and σ1 · · · σkσ′, for distinct σ and σ′. The rationale for keeping this second con-
junct is that with it, function symbols are annotated when used iff they are overloaded in this way. This means
that it is clear from the use of the function symbol, whether or not the annotation is required. This in turn
should help to improve human comprehension of scripts written using overloaded function symbols.

59 This is mostly a technical restriction, motivated by considerations of convenience. In fact, with a closed
formula ϕ of signature Σ the signature’s mapping of variables to sorts is irrelevant. The reason is that the
formula itself contains its own sort declaration for its term variables, either explicitly, for the variables bound
by a quantifier, or implicitly, for the variables bound by a let binder. Using only closed formulas then simpli-
fies the task of specifying their signature, as it becomes unnecessary to specify how the signature maps the
elements of X to the signature’s sorts.

60 Distinct sorts can have non-disjoint domain in a structure. However, whether they do that or not is irrelevant
in SMT-LIB logic. The reason is that the logic has no sort predicates, such as a subsort predicate, and does not
allow one to equate terms of different sorts (the term t1 ≈ t2 is ill-sorted unless t1 and t2 have the same sort).
As a consequence, a formula is satisfiable in a structure where two given sorts have non-disjoint domain iff it
is satisfiable in a structure where the two sorts do have disjoint domains.

61 This requirement is for concreteness. Again, since we work with closed formulas, which internally assign
sorts to their variables, the sorting of variables in a signature is irrelevant.

62 Admittedly, this requirement on theory declarations is somewhat hand-wavy. Unfortunately, it is not possible
to make it a lot more rigorous because a theory declaration can use natural language to define its class of
instance theories. The point is again that the definition of the class should impose no constraints on the
interpretation of free sort symbols and free function symbols.

Concrete Syntax

Predefined symbols

These symbols have a predefined meaning in Version 2.6. Note that they are not reserved words.
For instance, they could also be used in principle as user-defined sort or function symbols in
scripts.

Bool continued-execution error false immediate-exit incomplete logic memout sat
success theory true unknown unsupported unsat

Predefined keywords

These keywords have a predefined meaning in Version 2.6.

:all-statistics :assertion-stack-levels :authors :category :chainable :definition
:diagnostic-output-channel :error-behavior :extensions :funs :funs-description
:global-declarations :interactive-mode :language :left-assoc :license :name :named
:notes :pattern :print-success :produce-assignments :produce-models :produce-proofs
:produce-unsat-assumptions :produce-unsat-cores :random-seed :reason-unknown
:regular-output-channel :reproducible-resource-limit :right-assoc :smt-lib-version
:sorts :sorts-description :source :status :theories :values :verbosity :version

Auxiliary Lexical Categories

〈white_space_char〉 ::= 9dec | 10dec | 13dec | 32dec

〈printable_char〉 ::= 32dec | · · · | 126dec | 128dec | · · · | 255dec

〈digit〉 ::= 0 | · · · | 9

〈letter〉 ::= A | · · · | Z | a | · · · | z

Tokens

Reserved Words

General: ! _ as BINARY DECIMAL exists HEXADECIMAL forall let match NUMERAL
par STRING

Command names: assert check-sat check-sat-assuming declare-const
declare-datatype declare-datatypes declare-fun declare-sort define-fun
define-fun-rec define-sort echo exit get-assertions get-assignment get-info
get-model get-option get-proof get-unsat-assumptions get-unsat-core get-value
pop push reset reset-assertions set-info set-logic set-option

Other tokens

(

)
〈numeral〉 ::= 0 | a non-empty sequence of digits not starting with 0

〈decimal〉 ::= 〈numeral〉.0∗〈numeral〉
〈hexadecimal〉 ::= #x followed by a non-empty sequence of digits and letters

from A to F , capitalized or not

〈binary〉 ::= #b followed by a non-empty sequence of 0 and 1 characters

〈string〉 ::= sequence of whitespace and printable characters in double quotes
with escape sequence ""

〈simple_symbol〉 ::= a non-empty sequence of letters, digits and the characters
+ - / * = % ? ! . $ _ ˜ & ˆ < > @ that does not start
with a digit

〈symbol〉 ::= 〈simple_symbol〉
| a sequence of whitespace and printable characters that

starts and ends with | and does not otherwise include | or \
〈keyword〉 ::= :〈simple_symbol〉

Members of the 〈symbol〉 category starting with the character @ or . are reserved for solver
use. Solvers can use them respectively as identifiers for abstract values and solver generated
function symbols other than abstract values.

S-expressions

〈spec_constant〉 ::= 〈numeral〉 | 〈decimal〉 | 〈hexadecimal〉 | 〈binary〉 | 〈string〉
〈s_expr〉 ::= 〈spec_constant〉 | 〈symbol〉 | 〈keyword〉 | ( 〈s_expr〉∗ )

Identifiers

〈index〉 ::= 〈numeral〉 | 〈symbol〉
〈identifier〉 ::= 〈symbol〉 | ( _ 〈symbol〉 〈index〉+ )

Sorts

〈sort〉 ::= 〈identifier〉 | ( 〈identifier〉 〈sort〉+ )

Attributes

〈attribute_value〉 ::= 〈spec_constant〉 | 〈symbol〉 | ( 〈s_expr〉∗ )
〈attribute〉 ::= 〈keyword〉 | 〈keyword〉 〈attribute_value〉

Terms

〈qual_identifier〉 ::= 〈identifier〉 | ( as 〈identifier〉 〈sort〉 )
〈var_binding〉 ::= ( 〈symbol〉 〈term〉 )
〈sorted_var〉 ::= ( 〈symbol〉 〈sort〉 )
〈pattern〉 ::= 〈symbol〉 | ( 〈symbol〉 〈symbol〉+ )

〈match_case〉 ::= ( 〈pattern〉 〈term〉 )
〈term〉 ::= 〈spec_constant〉

| 〈qual_identifier〉
| ( 〈qual_identifier〉 〈term〉+ )
| ( let ( 〈var_binding〉+ ) 〈term〉 )
| ( forall ( 〈sorted_var〉+ ) 〈term〉 )
| ( exists ( 〈sorted_var〉+ ) 〈term〉 )
| ( match 〈term〉 ( 〈match_case〉+ ) )
| ( ! 〈term〉 〈attribute〉+ )

Theories

〈sort_symbol_decl〉 ::= ( 〈identifier〉 〈numeral〉 〈attribute〉∗ )
〈meta_spec_constant〉 ::= NUMERAL | DECIMAL | STRING

〈fun_symbol_decl〉 ::= ( 〈spec_constant〉 〈sort〉 〈attribute〉∗ )
| ( 〈meta_spec_constant〉 〈sort〉 〈attribute〉∗ )
| ( 〈identifier〉 〈sort〉+ 〈attribute〉∗ )

〈par_fun_symbol_decl〉 ::= 〈fun_symbol_decl〉
| ( par ( 〈symbol〉+ ) ( 〈identifier〉 〈sort〉+ 〈attribute〉∗ ) )

〈theory_attribute〉 ::= :sorts ( 〈sort_symbol_decl〉+ )
| :funs ( 〈par_fun_symbol_decl〉+ )
| :sorts-description 〈string〉
| :funs-description 〈string〉
| :definition 〈string〉
| :values 〈string〉
| :notes 〈string〉
| 〈attribute〉

〈theory_decl〉 ::= ( theory 〈symbol〉 〈theory_attribute〉+ )

Logics

〈logic_attribute〉 := :theories ( 〈symbol〉+ )
| :language 〈string〉
| :extensions 〈string〉
| :values 〈string〉
| :notes 〈string〉
| 〈attribute〉

〈logic〉 ::= ( logic 〈symbol〉 〈logic_attribute〉+ )

Info flags

〈info_flag〉 ::= :all-statistics | :assertion-stack-levels | :authors
| :error-behavior | :name | :reason-unknown
| :version | 〈keyword〉

Command options

〈b_value〉 ::= true | false

〈option〉 ::= :diagnostic-output-channel 〈string〉
| :global-declarations 〈b_value〉
| :interactive-mode 〈b_value〉
| :print-success 〈b_value〉
| :produce-assertions 〈b_value〉
| :produce-assignments 〈b_value〉
| :produce-models 〈b_value〉
| :produce-proofs 〈b_value〉
| :produce-unsat-assumptions 〈b_value〉
| :produce-unsat-cores 〈b_value〉
| :random-seed 〈numeral〉
| :regular-output-channel 〈string〉
| :reproducible-resource-limit 〈numeral〉
| :verbosity 〈numeral〉
| 〈attribute〉

Commands

〈sort_dec〉 ::= ( 〈symbol〉 〈numeral〉 )
〈selector_dec〉 ::= ( 〈symbol〉 〈sort〉 )
〈constructor_dec〉 ::= ( 〈symbol〉 〈selector_dec〉∗ )
〈datatype_dec〉 ::= ( 〈constructor_dec〉+ ) | ( par ( 〈symbol〉+ ) ( 〈constructor_dec〉+ ) )

〈function_dec〉 ::= ( 〈symbol〉 ( 〈sorted_var〉∗ ) 〈sort〉 )
〈function_def 〉 ::= 〈symbol〉 ( 〈sorted_var〉∗ ) 〈sort〉 〈term〉
〈prop_literal〉 ::= 〈symbol〉 | ( not 〈symbol〉 )
〈command〉 ::= ( assert 〈term〉 )

| ( check-sat )
| ( check-sat-assuming ( 〈prop_literal〉∗ ) )
| ( declare-const 〈symbol〉 〈sort〉 )
| ( declare-datatype 〈symbol〉 〈datatype_dec〉)
| ( declare-datatypes ( 〈sort_dec〉n+1 ) ( 〈datatype_dec〉n+1 ) )
| ( declare-fun 〈symbol〉 ( 〈sort〉∗ ) 〈sort〉 )
| ( declare-sort 〈symbol〉 〈numeral〉 )
| ( define-fun 〈function_def 〉 )
| ( define-fun-rec 〈function_def 〉 )
| ( define-funs-rec ( 〈function_dec〉n+1 ) ( 〈term〉n+1 ) )
| ( define-sort 〈symbol〉 ( 〈symbol〉∗ ) 〈sort〉 )
| ( echo 〈string〉 )
| ( exit )
| ( get-assertions )
| ( get-assignment )
| ( get-info 〈info_flag〉 )
| ( get-model )
| ( get-option 〈keyword〉 )
| ( get-proof )
| ( get-unsat-assumptions )
| ( get-unsat-core )
| ( get-value ( 〈term〉+ ) )
| ( pop 〈numeral〉 )
| ( push 〈numeral〉 )
| ( reset )
| ( reset-assertions )
| ( set-info 〈attribute〉 )
| ( set-logic 〈symbol〉 )
| ( set-option 〈option〉 )

〈script〉 ::= 〈command〉∗

Command responses

〈error-behavior〉 ::= immediate-exit | continued-execution
〈reason-unknown〉 ::= memout | incomplete | 〈s_expr〉
〈model_response〉 ::= ( define-fun 〈function_def 〉 ) | ( define-fun-rec 〈function_def 〉 )

| ( define-funs-rec ( 〈function_dec〉n+1 ) ( 〈term〉n+1 ) )

〈info_response〉 ::= :assertion-stack-levels 〈numeral〉
| :authors 〈string〉
| :error-behavior 〈error-behavior〉
| :name 〈string〉
| :reason-unknown 〈reason-unknown〉
| :version 〈string〉
| 〈attribute〉

〈valuation_pair〉 ::= ( 〈term〉 〈term〉 )
〈t_valuation_pair〉 ::= ( 〈symbol〉 〈b_value〉 )
〈check_sat_response〉 ::= sat | unsat | unknown

〈echo_response〉 ::= 〈string〉
〈get_assertions_response〉 ::= ( 〈term〉∗ )
〈get_assignment_response〉 ::= ( 〈t_valuation_pair〉∗ )
〈get_info_response〉 ::= ( 〈info_response〉+ )

〈get_model_response〉 ::= ( 〈model_response〉∗ )
〈get_option_response〉 ::= 〈attribute_value〉
〈get_proof_response〉 ::= 〈s_expr〉
〈get_unsat_assump_response〉 ::= ( 〈symbol〉∗ )
〈get_unsat_core_response〉 ::= ( 〈symbol〉∗ )
〈get_value_response〉 ::= ( 〈valuation_pair〉+ )

〈specific_success_response〉 ::= 〈check_sat_response〉 | 〈echo_response〉
| 〈get_assertions_response〉 | 〈get_assignment_response〉
| 〈get_info_response〉 | 〈get_model_response〉
| 〈get_option_response〉 | 〈get_proof_response〉
| 〈get_unsat_assumptions_response〉
| 〈get_unsat_core_response〉 | 〈get_value_response〉

〈general_response〉 ::= success | 〈specific_success_response〉
| unsupported | ( error 〈string〉 )

Abstract Syntax

Common Notation

b ∈ B, the set of boolean values r ∈ R, the set of non-negative rational numbers
n ∈ N , the set of natural numbers w ∈ W , the set of character strings
s ∈ S , the set of sort symbols u ∈ U , the set of sort parameters
f ∈ F , the set of function symbols x ∈ X , the set of variables
a ∈ A, the set of attribute names v ∈ V , the set of attribute values
T ∈ T N , the set of theory names L ∈ L, the set of logic names

Sorts

(Sorts) σ ::= s σ∗

(Parametric Sorts) τ ::= u | s τ∗

Terms

(Attributes) α ::= a | a = v

(Patterns) p ::= x | f x∗

(Terms) t ::= x | f t∗ | f σ t∗

| ∃ (x:σ)+ t | ∀ (x:σ)+ t | let (x = t)+ in t | match t with (p→ t)+

| t α+

Well-sorting rules for terms

Σ ` x α∗ : σ
if x:σ ∈ Σ

Σ ` t1 : σ1 · · · Σ ` tk : σk

Σ ` ( f t1 · · · tk) α∗ : σ
if

{
f :σ1 · · · σkσ ∈ Σ,
f :σ1 · · · σkσ′ /∈ Σ for all σ′ 6= σ

Σ ` t1 : σ1 · · · Σ ` tk : σk

Σ ` ( f σ t1 · · · tk) α∗ : σ
if

{
f :σ1 · · · σkσ ∈ Σ,
f :σ1 · · · σkσ′ ∈ Σ for some σ′ 6= σ

Σ[x1:σ1, . . . , xk+1:σk+1] ` t : Bool
Σ ` (Qx1:σ1 · · · xk+1:σk+1 t) α∗ : Bool

if Q ∈ {∃, ∀}

Σ ` t1 : σ1 · · · Σ ` tk+1 : σk+1

Σ[x1:σ1, . . . , xk+1:σk+1] ` t : σ

Σ ` (let x1 = t1 · · · xk+1 = tk+1 in t) α∗ : σ

if x1, . . . , xk+1 are all distinct

Σ ` t : δ

Σ[x̄i :σ̄i] ` ti : σ for i = 1, . . . , k + 1

Σ ` (match t with c1 x̄1 → t1 · · · ck+1 x̄k+1 → tk+1) α∗ : σ

if


{c1, . . . , ck+1} = conΣ(δ),
for all i = 1, . . . , k + 1

ci :σ̄iδ ∈ Σ and
x̄i contains no repetitions

Σ ` t : δ

Σ[x̄i :σ̄i] ` ti : σ for i = 1, . . . , k
Σ[xi :δ] ` ti : σ for i = k + 1, . . . , n

Σ ` (match t with p1 → t1 · · · pn → tn) α∗ : σ

if



{c1, . . . , ck} ⊆ conΣ(δ) 6= ∅,
{p1, . . . , pn} = {c1 x̄1, . . . , ck x̄k} ∪

{xk+1, . . . , xn},
k < n,
for all i = 1, . . . , k

ci :σ̄iδ ∈ Σ and
x̄i contains no repetitions

Theories

(Sort symbol declarations) sdec ::= s n α∗

(Fun. symbol declarations) fdec ::= f σ+ α∗

(Param. fun. symbol declarations) pdec ::= fdec | Π u+ ( f τ+ α∗)

(Theory attributes) tattr ::= sorts = sdec+ | funs = pdec+

| sorts-description = w
| funs-description = w
| definition = w | values = t+

| notes = w | α

(Theory declarations) tdec ::= theory T tattr+



103

Logics

(Logic attributes) lattr ::= theories = T+ | language = w
| extensions = w | values = w
| notes = w | α

(Logic declarations) ldec ::= logic L lattr+

